
Troubleshooting
Querying the Jobs Table
Oban.Job defines an Ecto schema and the jobs table can therefore be queried as usual, e.g.:
MyApp.Repo.all(
  from j in Oban.Job,
    where: j.worker == "MyApp.Business",
    where: j.state == "discarded"
)
Heroku
Elixir and Erlang versions
If your app crashes on launch, be sure to confirm you are running the correct
version of Elixir and Erlang (view requirements). If using the
hashnuke/elixir buildpack, you can update the elixir_buildpack.config file
in your application's root directory to something like:
# Elixir version
elixir_version=1.9.0

# Erlang version
erlang_version=22.0.3
Available Erlang versions are available here.
Database connections
Make sure that you have enough available database connections when running on
Heroku. Oban uses a database connection in order to listen for pubsub
notifications. This is in addition to your Ecto Repo pool_size setting.
Heroku's Hobby tier Postgres plans
have a maximum of 20 connections, so if you're using one of those plan
accordingly.
  
Release Configuration
While having the same Oban configuration for every environment might be fine,
there are certainly times you might want to make changes for a specific
environment. For example, you may want to increase or decrease a queue's
concurrency.
Using Config Providers
If you are using Elixir Releases, this is straight forward to do using Module
Config Providers:
defmodule MyApp.ConfigProvider do
  @moduledoc """
  Provide release configuration for Oban Queue Concurrency
  """

  @behaviour Config.Provider

  def init(path) when is_binary(path), do: path

  def load(config, path) do
    case parse_json(path) do
      nil ->
        config

      queues ->
        Config.Reader.merge(config, ingestion: [{Oban, [queues: queues]}])
    end
  end

  defp parse_json(path) do
    {:ok, _} = Application.ensure_all_started(:jason)

    if File.exists?(path) do
      path
      |> File.read!()
      |> Jason.decode!()
      |> Map.fetch!("queues")
      |> Keyword.new(fn {key, value} -> {String.to_atom(key), value} end)
    end
  end
end
Our config provider ensures that the Jason app is loaded so that we can parse
a JSON configuration file. Once the JSON is loaded we must extract the
queues map and convert it to a keyword list where all of the keys are atoms.
The use of String.to_atom/1 is safe because all of our queues names are
already defined.
Then you include this in your mix.ex file, where your release is configured:
releases: [
  umbrella_app: [
    version: "0.0.1",
    applications: [
      child_app: :permanent
    ],
    config_providers: [{Path.To.ConfigProvider, "/etc/config.json"}]
  ]
]
Then when you release your app, you ensure that you have a JSON file mounted at
whatever path you specified above and that it contains all of your desired queues:
{"queues": {"special": 1, "default": 10, "events": 20}}
  
Recursive Jobs
Recursive jobs, like recursive functions, call themselves after they have have
executed. Except unlike recursive functions, where recursion happens in a tight
loop, a recursive job enqueues a new version of itself and may add a slight
delay to alleviate pressure on the queue.
Recursive jobs are a great way to backfill large amounts of data where a
database migration or a mix task may not be suitable. Here are a few reasons
that a recursive job may be better suited for backfilling data:
	Data can't be backfilled with a database migration, it may require talking to
an external service
	A task may fail partway through execution; resuming the task would mean
starting over again, or tracking progress manually to resume where the failure
occurred
	A task may be computationally intensive or put heavy pressure on the database
	A task may run for too long and would be interrupted by code releases or other
node restarts
	A task may interface with an external service and require some rate limiting
	A job can be used directly for new records and to backfill existing records

Let's explore recursive jobs with a use case that builds on several of those
reasons.
Use Case: Backfilling Timezone Data
Consider a worker that queries an external service to determine what timezone a
user resides in. The external service has a rate limit and the response time is
unpredictable. We have a lot of users in our database missing timezone
information, and we need to backfill.
Our application has an existing TimezoneWorker that accepts a user's id,
makes an external request and then updates the user's timezone. We can modify
the worker to handle backfilling by adding a new clause to perform/1. The new
clause explicitly checks for a backfill argument and will enqueue the next job
after it executes:
defmodule MyApp.Workers.TimezoneWorker do
  use Oban.Worker

  import Ecto.Query

  alias MyApp.{Repo, User}

  @backfill_delay 1

  @impl true
  def perform(%{args: %{"id" => id, "backfill" => true}}) do
    with :ok <- perform(%{args: %{"id" => id}}) do
      case fetch_next(id) do
        next_id when is_integer(next_id) ->
          %{id: next_id, backfill: true}
          |> new(schedule_in: @backfill_delay)
          |> Oban.insert!()

        nil ->
          :ok
      end
    end
  end

  def perform(%{args: %{"id" => id}}) do
    update_timezone(id)
  end

  defp fetch_next(current_id) do
    User
    |> where([u], is_nil(u.timezone))
    |> where([u], u.id > ^current_id)
    |> order_by(asc: :id)
    |> limit(1)
    |> select([u], u.id)
    |> Repo.one()
  end

  defp update_timezone(_id), do: Enum.random([:ok, {:error, :reason}])
end
There is a lot happening in the worker module, so let's unpack it a little bit.
	There are two clauses for perform/1, the first only matches when a job is
marked as "backfill" => true, the second does the actual work of updating the
timezone.
	The backfill clause checks that the timezone update succeeds and then uses
fetch_next/1 to look for the id of the next user without a timezone.
	When another user needing a backfill is available it enqueues a new backfill
job with a one second delay.

With the new perform/1 clause in place and our code deployed we can kick off
the recursive backfill. Assuming the id of the first user is 1, you can
start the job from an iex console:
iex> %{id: 1, backfill: true} |> MyApp.Workers.TimezoneWorker.new() |> Oban.insert()
Now the jobs will chug along at a steady rate of one per second until the
backfill is complete (or something fails). If there are any errors the backfill
will pause until the failing job completes: especially useful for jobs relying
on flaky external services. Finally, when there aren't any more user's without a
timezone, the backfill is complete and recursion will stop.
Building On Recursive Jobs
This was a relatively simple example, and hopefully it illustrates the power and
flexibility of recursive jobs. Recursive jobs are a general pattern and aren't
specific to Oban. In fact, aside from the use Oban.Worker directive there
isn't anything specific to Oban in the recipe!
  
Reliable Scheduled Jobs
A common variant of recursive jobs are "scheduled jobs", where the goal is for a
job to repeat indefinitely with a fixed amount of time between executions. The
part that makes it "reliable" is the guarantee that we'll keep retrying the
job's business logic when the job retries, but we'll only schedule the next
occurrence once. In order to achieve this guarantee we'll make use of the
perform function to receive a complete Oban.Job struct.
Time for illustrative example!
Use Case: Delivering Daily Digest Emails
When a new user signs up to use our site we need to start sending them daily
digest emails. We want to deliver the emails around the same time a user signed
up every, repeating every 24 hours. It is important that we don't spam them with
duplicate emails, so we ensure that the next email is only scheduled on our
first attempt.
defmodule MyApp.Workers.ScheduledWorker do
  use Oban.Worker, queue: :scheduled, max_attempts: 10

  alias MyApp.Mailer

  @one_day 60 * 60 * 24

  @impl true
  def perform(%{args: %{"email" => email} = args, attempt: 1}) do
    args
    |> new(schedule_in: @one_day)
    |> Oban.insert!()

    Mailer.deliver_email(email)
  end

  def perform(%{args: %{"email" => email}}) do
    Mailer.deliver_email(email)
  end
end
You'll notice that the first perform/1 clause only matches a job struct on the
first attempt. When it matches, the first clause schedules the next iteration
immediately, before attempting to deliver the email. Any subsequent retries
fall through to the second perform/1 clause, which only attempts to deliver
the email again. Combined, the clauses get us close to at-most-once semantics
for scheduling, and at-least-once semantics for delivery.
More Flexible Than CRON Scheduling
Delivering around the same time using cron-style scheduling would need extra
book-keeping to check when a user signed up, and then only deliver to those
users that signed up within that window of time. The recursive scheduling
approach is more accurate and entirely self contained—when and if the digest
interval changes the scheduling will pick it up automatically once our code
deploys.
An extensive discussion on the Oban issue tracker prompted this example
along with the underlying feature that made it possible.
  
Reporting Job Progress
Most applications provide some way to generate an artifact—something that may
take the server a long time to accomplish. If it takes several minutes to render
a video, crunch some numbers or generate an export, users may be left wondering
whether your application is working. Providing periodic updates to end users
assures them that the work is being done and keeps the application feeling
responsive.
Reporting progress is something that any background job processor with
unlimited execution time can do! Naturally, we'll look at an example built on
Oban.
Use Case: Exporting a Large Zip File
Users of our site can export a zip of all the files they have uploaded. A zip
file (no, not a tar, our users don't have neck-beards) is generated on the fly,
when the user requests it. Lazily generating archives is great for our server's
utilization, but it means that users may wait a while when there are many files.
Fortunately, we know how many files will be included in the zip and we can use
that information to send progress reports! We will compute the archive's percent
complete as each file is added and push a message to the user.
Before We Start
In the forum question that prompted this guide the work was done
externally by a port process. Working with ports is well outside the scope of
this guide, so I've modified it for the sake of simplicity. The result is
slightly contrived as it puts both processes within the same module, which isn't
necessary if the only goal is to broadcast progress. This guide is ultimately
about coordinating processes to report progress from a background job, so that's
what we'll focus on (everything else will be rather hand-wavy).
Coordinating Processes
Our worker, the creatively titled ZippingWorker, handles both building the
archive and reporting progress to the client. Showing the entire module at once
felt distracting, so we'll start with only the module definition and the
perform/1 function:
defmodule MyApp.Workers.ZippingWorker do
  use Oban.Worker, queue: :exports, max_attempts: 1

  alias MyApp.{Endpoint, Zipper}

  def perform(%_{args: %{"channel" => channel, "paths" => paths}}) do
    build_zip(paths)
    await_zip(channel)
  end

  # ...
end
The function accepts an Oban Job with a channel name and a list of file paths,
which it immediately passes on to the private build_zip/1:
  defp build_zip(paths) do
    job_pid = self()

    Task.async(fn ->
      zip_path = Zipper.new()

      paths
      |> Enum.with_index(1)
      |> Enum.each(fn {path, index} ->
        :ok = Zipper.add_file(zip_path, path)
        send(job_pid, {:progress, trunc(index / length(paths) * 100)})
      end)

      send(job_pid, {:complete, zip_path})
    end)
  end
The function grabs the current pid, which belongs to the job, and kicks off an
asynchronous task to handle the zipping. With a few calls to a fictional
Zipper module the task works through each file path, adding it to the zip.
After adding a file the task sends a :progress message with the percent
complete back to the job. Finally, when the zip finishes, the task sends a
:complete message with a path to the archive.
The asynchronous call spawns a separate process and returns immediately. In
order for the task to finish building the zip we need to wait on it. Typically
we'd use Task.await/2, but we'll use a custom receive loop to track the task's
progress:
  defp await_zip(channel) do
    receive do
      {:progress, percent} ->
        Endpoint.broadcast(channel, "zip:progress", percent)
        await_zip(channel)

      {:complete, zip_path} ->
        Endpoint.broadcast(channel, "zip:complete", zip_path)
    after
      30_000 ->
        Endpoint.broadcast(channel, "zip:failed", "zipping failed")
        raise RuntimeError, "no progress after 30s"
    end
  end
The receive loop blocks execution while it waits for :progress or :complete
messages. When a message comes in it broadcasts to the provided channel and the
client receives an update (this example uses Phoenix Channels, but any
other PubSub type mechanism would work). As a safety mechanism we have an
after clause that will timeout after 30 seconds of inactivity. If the receive
block times out we notify the client and raise an error, failing the job.
Made Possible by Unlimited Execution
Reporting progress asynchronously works in Oban because anything that blocks a
worker's perform/1 function will keep the job executing. Jobs aren't executed
inside of a transaction, which alleviates any limitations on how long a job can
run.
This technique is suitable for any single long running job where an end user
is waiting on the results. You can read the "Batching Jobs for Monitoring" guide
if you need to combine multiple jobs into a single output by creating batch
jobs.
  
Batching Jobs for Monitoring
In the Reporting Progress guide we looked at tracking the
progress of a single job as it executes. What about tracking the progress of
tens, hundreds or thousands of jobs as they execute? In that situation we want
to monitor the jobs as a group—execute them in parallel and then enqueue a
callback when all the jobs are finished. At least one popular background job
processor calls these groups "batches", and so we'll adopt that term here
as we build it out with Oban.
Use Case: Notifying Admins When an Email Delivery is Complete
Admins on our site send weekly batch emails to a large mailing list to let users
know new content is available. Naturally the system sends emails in parallel in
the background. Delivery can take many hours and we want to notify our admins
when the batch is complete. This is an admittedly simple use case, but it is
just complex enough to benefit from a batching flow.
At a high level, the worker flow looks like this:
	Generate a unique id for the batch, it can be entirely random or something
structured like "my-batch-1"; any string will due, provided it is unique for
the forseable future.
	Count the total number of jobs to execute. This is the batch_size, which
we'll use later to decide whether all jobs have completed.
	Create a worker that has a perform/1 clause matching a batch_id key. This
clause will handle the real work that the job is meant to do, and afterwards
it will start a separate process to check whether the batch is complete.
Since executed jobs are stored in the database with a completed state we
can evaluate whether this was the final job in the batch.
	When we detect that the current job is the last one we enqueue one final job
with different arguments to indicate that it is the "completed" callback.
Through the magic of pattern matching this "callback" job can live within the
same worker.

Here is the worker module with both the primary and callback clauses of
perform/1:
defmodule MyApp.Workers.BatchEmailWorker do
  use Oban.Worker, queue: :batch, unique: [period: 60]

  import Ecto.Query

  @final_check_delay 50

  alias MyApp.{Mailer, Repo}

  @impl true
  def perform(%{args: %{"email" => email, "batch_id" => batch_id, "batch_size" => batch_size}}) do
    Mailer.weekly_update(email)

    Task.start(fn ->
      Process.sleep(@final_check_delay)

      if final_batch_job?(batch_id, batch_size) do
        %{"status" => "complete", "batch_id" => batch_id}
        |> new()
        |> Oban.insert()
      end
    end)
  end

  def perform(%{args: {"status" => "complete", "batch_id" => batch_id}}) do
    Mailer.notify_admin("Batch #{batch_id} is complete!")
  end
end
Within the first perform/1 clause we deliver a weekly update email and then
start a separate task to check whether this is the final job. The task is not
linked to the job and it uses a short sleep to give enough time for the job to
be marked complete; the goal is to prevent race conditions where no callback is
ever enqueued. The final_batch_job?/2 function is wrapper around a fairly
involved Ecto query:
defp final_batch_job?(batch_id, batch_size) do
  Oban.Job
  |> where([j], j.state not in ["available", "executing", "scheduled"])
  |> where([j], j.queue == "batch")
  |> where([j], fragment("?->>'batch_id' = ?", j.args, ^batch_id))
  |> where([j], not fragment("? \\? 'status'", j.args))
  |> select([j], count(j.id) >= ^batch_size)
  |> Repo.one()
end
This private predicate function uses the Oban.Job struct to query the
oban_jobs table for other completed jobs in the batch. Within the query we use
a fragment containing the indecipherable ->> operator, a native PostgreSQL
jsonb operator that keys into the args column and filters down to jobs
in the same batch. The equally indecipherable existence operator (\\?), which
must be double escaped within a fragment, helps to ensure that we aren't
creating duplicate callback jobs. When the number of completed or discarded jobs
matches our expected batch size we know that the batch is complete!
It's worth mentioning at this point that by default there aren't any indexes on
the args column, so this query won't be super snappy with a lot of completed
jobs laying around. If you plan on integrating batches into your workflow, and
you want to ensure that callback jobs are absolutely unique, you should add
a unique index on batch_id, and possibly one for the status argument.
To kick off our batch job we generate a batch_id and a iterate through a list
of emails:
batch_id = "email-blast-#{DateTime.to_unix(DateTime.utc_now())}"
batch_size = length(emails)

for email <- emails do
  %{email: email, batch_id: batch_id, batch_size: batch_size}
  |> Oban.Workers.BatchEmailWorker.new()
  |> Oban.insert!()
end
Historic Observation
This batching technique is possible without any other tables or tracking
mechanisms because Oban's jobs are retained in the database after execution.
They're stored right along with your other production data, which opens them up
to querying and manipulating as needed. Batching isn't built into Oban because
between queries and pattern matching you have everything you need to build
complex batch pipelines.
One final note: querying for completed batches all hinges on how aggressive your
pruning configuration is. If you're pruning completed jobs after a few minutes
or a few hours then there is a good chance that your batch won't ever complete.
Be sure that you tune your pruning so that there is enough headroom for
batches to finish.
  
Handling Expected Failures
Reporting job errors by sending notifications to an external service is
essential to maintaining application health. While reporting is essential, noisy
reports for flaky jobs can become a distraction that gets ignored. Sometimes we
expect that a job will error a few times. That could be because the job relies
on an external service that is flaky, because it is prone to race conditions, or
because the world is a crazy place. Regardless of why a job fails, reporting
every failure may be undesirable.
Use Case: Silencing Initial Notifications for Flaky Services
One solution for reducing noisy error notifications is to start reporting only
after a job has failed several times. Oban uses Telemetry to make
reporting errors and exceptions a simple matter of attaching a handler function.
In this example we will extend Honeybadger reporting from the
Oban.Telemetry documentation, but account for the number of processing attempts.
To start, we'll define a Reportable protocol with a single
reportable?/2 function:
defprotocol MyApp.Reportable do
  @fallback_to_any true
  def reportable?(worker, attempt)
end

defimpl MyApp.Reportable, for: Any do
  def reportable?(_worker, _attempt), do: true
end
The Reportable protocol has a default implementation which always returns
true, meaning it reports all errors. Our application has a FlakyWorker
that's known to fail a few times before succeeding. We don't want to see a
report until after a job has failed three times, so we'll add an implementation
of Reportable within the worker module:
defmodule MyApp.Workers.FlakyWorker do
  use Oban.Worker

  defimpl MyApp.Reportable do
    @threshold 3

    def reportable?(_worker, attempt), do: attempt > @threshold
  end

  @impl true
  def perform(%{args: %{"email" => email}}) do
    MyApp.ExternalService.deliver(email)
  end
end
The final step is to call reportable?/2 from our application's error reporter,
passing in the worker module and the attempt number:
defmodule MyApp.ErrorReporter do
  alias MyApp.Reportable

  def handle_event(_, _, %{attempt: attempt, worker: worker} = meta, _) do
    if Reportable.reportable?(worker, attempt) do
      context = Map.take(meta, [:id, :args, :queue, :worker])

      Honeybadger.notify(meta.error, context, meta.stack)
    end
  end
end
Attach the failure handler somewhere in your application.ex module:
:telemetry.attach("oban-errors", [:oban, :job, :exception], &ErrorReporter.handle_event/4, nil)
With the failure handler attached you will start getting error reports only
after the third error.
Giving Time to Recover
If a service is especially flaky you may find that Oban's default backoff
strategy is too fast. By defining a custom backoff function on the
FlakyWorker we can set a linear delay before retries:
# inside of MyApp.Workers.FlakyWorker

@impl true
def backoff(attempt, base_amount \\ 60) do
  attempt * base_amount
end
Now the first retry is scheduled 60s later, the second 120s later, and so on.
Building Blocks
Elixir's powerful primitives of behaviours, protocols and event handling make
flexible error reporting seamless and extendible. While our Reportable
protocol only considered the number of attempts, this same mechanism is suitable
for filtering by any other meta value.
Explore the [event metadata][meta] that Oban provides for job failures to see
how you can configure reporting by by worker, queue, or even specific arguments.
  
Splitting Queues Between Nodes
Running every job queue on every node isn't always ideal. Imagine that your
application has some CPU intensive jobs that you'd prefer not to run on nodes
that serve web requests. Perhaps you start temporary nodes that are only meant
to insert jobs but should never execute any. Fortunately, we can control
this by configuring certain node types, or even single nodes, to run only a
subset of queues.
Use Case: Isolating Video Processing Intensive Jobs
One notorious type of CPU intensive work is video processing. When our
application is transcoding multiple videos simultaneously it is a major drain
on system resources and may impact response times. To avoid this we can run
dedicated worker nodes that don't serve any web requests and handle all of the
transcoding.
While it's possible to separate our system into web and worker apps within
an umbrella, that wouldn't allow us to dynamically change queues at runtime.
Let's look at an environment variable based method for dynamically
configuring queues at runtime.
Within config.exs our application is configured to run three queues:
default, media and events:
config :my_app, Oban,
  repo: MyApp.Repo,
  queues: [default: 15, media: 10, events: 25]
We will use an OBAN_QUEUES environment variable to override the queues at
runtime. For illustration purposes the queue parsing all happens within the
application module, but it would work equally well in releases.exs.
defmodule MyApp.Application do
  @moduledoc false

  use Application

  def start(_type, _args) do
    children = [
      MyApp.Repo,
      MyApp.Endpoint,
      {Oban, oban_opts()}
    ]

    Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
  end

  defp oban_opts do
    env_queues = System.get_env("OBAN_QUEUES")

    :my_app
    |> Application.get_env(Oban)
    |> Keyword.update(:queues, [], &queues(env_queues, &1))
  end

  defp queues("*", defaults), do: defaults
  defp queues(nil, defaults), do: defaults
  defp queues(false, _), do: false

  defp queues(values, _defaults) when is_binary(values) do
    values
    |> String.split(" ", trim: true)
    |> Enum.map(&String.split(&1, ",", trim: true))
    |> Keyword.new(fn [queue, limit] ->
      {String.to_existing_atom(queue), String.to_integer(limit)}
    end)
  end
end
The queues function's first three clauses ensure that we can fall back to the
queues specified in our configuration (or false, for testing). The fourth
clause is much more involved, and that is where the environment parsing happens.
It expects the OBAN_QUEUES value to be a string formatted as queue,limit
pairs and separated by spaces. For example, to run only the default and
media queues with a limit of 10 and 5 respectively, you would pass the string
default,5 media,10.
Note that the parsing clause has a couple of safety mechanisms to ensure that
only real queues are specified:
	It automatically trims while splitting values, so extra whitespace like won't
break parsing (i.e. default,3)
	It only converts the queue string to an existing atom, hopefully
preventing typos that would start a random queue (i.e. defalt)

Usage Examples
In development (or when using mix rather than releases) we can specify the
environment variable inline:
OBAN_QUEUES="default,10 media,5" mix phx.server # default: 10, media: 5
We can also explicitly opt in to running all of the configured queues:
OBAN_QUEUES="*" mix phx.server # default: 15, media: 10, events: 25
Finally, without OBAN_QUEUES set at all it will implicitly fall back to the
configured queues:
mix phx.server # default: 15, media: 10, events: 25
Flexible Across all Environments
This environment variable based solution is more flexible than running separate
umbrella apps because we can reconfigure at any time. In a limited environment,
like staging, we can run all the queues on a single node using the exact same
code we use in production. In the future, if other workers start to utilize too
much CPU or RAM we can shift them to the worker node without any code
changes.
This guide was prompted by an inquiry on the Oban issue tracker.

Note: This README is for the unreleased master branch, please reference the
official documentation on hexdocs for the latest stable release.

Features
Oban's primary goals are reliability, consistency and observability.
It is fundamentally different from other background job processing tools because
it retains job data for historic metrics and inspection. You can leave your
application running indefinitely without worrying about jobs being lost or
orphaned due to crashes.
Advantages Over Other Tools
	Fewer Dependencies — If you are running a web app there is a very good
chance that you're running on top of a RDBMS. Running your job queue
within PostgreSQL minimizes system dependencies and simplifies data backups.

	Transactional Control — Enqueue a job along with other database changes,
ensuring that everything is committed or rolled back atomically.

	Database Backups — Jobs are stored inside of your primary database, which
means they are backed up together with the data that they relate to.


Advanced Features
	Isolated Queues — Jobs are stored in a single table but are executed in
distinct queues. Each queue runs in isolation, ensuring that a job in a single
slow queue can't back up other faster queues.

	Queue Control — Queues can be started, stopped, paused, resumed and scaled
independently at runtime across all running nodes (even in environments like
Heroku, without distributed Erlang).

	Resilient Queues — Failing queries won't crash the entire supervision tree,
instead they trip a circuit breaker and will be retried again in the future.

	Job Canceling — Jobs can be canceled in the middle of execution regardless
of which node they are running on. This stops the job at once and flags it as
discarded.

	Triggered Execution — Database triggers ensure that jobs are dispatched as
soon as they are inserted into the database.

	Unique Jobs — Duplicate work can be avoided through unique job controls.
Uniqueness can be enforced at the argument, queue and worker level for any
period of time.

	Scheduled Jobs — Jobs can be scheduled at any time in the future, down to
the second.

	Periodic (CRON) Jobs — Automatically enqueue jobs on a cron-like schedule.
Duplicate jobs are never enqueued, no matter how many nodes you're running.

	Job Priority — Prioritize jobs within a queue to run ahead of others.

	Historic Metrics — After a job is processed the row isn't deleted.
Instead, the job is retained in the database to provide metrics. This allows
users to inspect historic jobs and to see aggregate data at the job, queue or
argument level.

	Node Metrics — Every queue records metrics to the database during runtime.
These are used to monitor queue health across nodes and may be used for
analytics.

	Queue Draining — Queue shutdown is delayed so that slow jobs can finish
executing before shutdown. When shutdown starts queues are paused and stop
executing new jobs. Any jobs left running after the shutdown grace period may
be rescued later.

	Telemetry Integration — Job life-cycle events are emitted via
Telemetry integration. This enables simple logging, error reporting
and health checkups without plug-ins.


Requirements
Oban has been developed and actively tested with Elixir 1.8+, Erlang/OTP 21.1+
and PostgreSQL 11.0+. Running Oban currently requires Elixir 1.8+, Erlang 21+,
and PostgreSQL 9.6+.
Oban Web UI
A web-based user interface for monitoring and managing Oban is available as a
private package. Learn more about the UI at oban.dev.
Installation
Oban is published on Hex. Add it to your list of
dependencies in mix.exs:
def deps do
  [
    {:oban, "~> 2.0.0-rc.1"}
  ]
end
Then run mix deps.get to install Oban and its dependencies, including
Ecto, Jason and Postgrex.
After the packages are installed you must create a database migration to
add the oban_jobs table to your database:
mix ecto.gen.migration add_oban_jobs_table
Open the generated migration in your editor and call the up and down
functions on Oban.Migrations:
defmodule MyApp.Repo.Migrations.AddObanJobsTable do
  use Ecto.Migration

  def up do
    Oban.Migrations.up()
  end

  # We specify `version: 1` in `down`, ensuring that we'll roll all the way back down if
  # necessary, regardless of which version we've migrated `up` to.
  def down do
    Oban.Migrations.down(version: 1)
  end
end
This will run all of Oban's versioned migrations for your database. Migrations
between versions are idempotent and will never change after a release. As new
versions are released you may need to run additional migrations.
Now, run the migration to create the table:
mix ecto.migrate
Next see Usage for how to integrate Oban into your application and
start defining jobs!
Note About Releases
If you are using releases you may see Postgrex errors logged during your initial
deploy (or any deploy requiring an Oban migration). The errors are only
temporary! After the migration has completed each queue will start producing
jobs normally.
Usage
Oban is a robust job processing library which uses PostgreSQL for storage and
coordination.
Each Oban instance is a supervision tree and not an application. That means it
won't be started automatically and must be included in your application's
supervision tree. All of your configuration is passed into the supervisor,
allowing you to configure Oban like the rest of your application:
# config/config.exs
config :my_app, Oban,
  repo: MyApp.Repo,
  queues: [default: 10, events: 50, media: 20]

# lib/my_app/application.ex
defmodule MyApp.Application do
  @moduledoc false

  use Application

  alias MyApp.Repo
  alias MyAppWeb.Endpoint

  def start(_type, _args) do
    children = [
      Repo,
      Endpoint,
      {Oban, oban_config()}
    ]

    Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
  end

  defp oban_config do
    opts = Application.get_env(:my_app, Oban)

    # Prevent running queues or scheduling jobs from an iex console, i.e. when starting app with `iex -S mix`
    if Code.ensure_loaded?(IEx) and IEx.started?() do
      opts
      |> Keyword.put(:crontab, false)
      |> Keyword.put(:queues, false)
    else
      opts
    end
  end
end
If you are running tests (which you should be) you'll want to disable pruning
, enqueuing scheduled jobs and job dispatching altogether when testing:
# config/test.exs
config :my_app, Oban, crontab: false, queues: false, plugins: false
Configuring Queues
Queues are specified as a keyword list where the key is the name of the queue
and the value is the maximum number of concurrent jobs. The following
configuration would start four queues with concurrency ranging from 5 to 50:
queues: [default: 10, mailers: 20, events: 50, media: 5]
There isn't a limit to the number of queues or how many jobs may execute
concurrently in each queue. Here are a few caveats and guidelines:
Caveats & Guidelines
	Each queue will run as many jobs as possible concurrently, up to the
configured limit. Make sure your system has enough resources (i.e. database
connections) to handle the concurrent load.

	Queue limits are local (per-node), not global (per-cluster). For example,
running a queue with a local limit of one on three separate nodes is
effectively a global limit of three. If you require a global limit you must
restrict the number of nodes running a particular queue.

	Only jobs in the configured queues will execute. Jobs in any other queue will
stay in the database untouched.

	Be careful how many concurrent jobs make expensive system calls (i.e. FFMpeg,
ImageMagick). The BEAM ensures that the system stays responsive under load,
but those guarantees don't apply when using ports or shelling out commands.


Defining Workers
Worker modules do the work of processing a job. At a minimum they must define a
perform/2 function, which is called with an args map and the job struct.
Note that the args map passed to perform/2 will always have string keys,
regardless of the key type when the job was enqueued. The args are stored as
jsonb in PostgreSQL and the serialization process automatically stringifies
all keys.
Define a worker to process jobs in the events queue:
defmodule MyApp.Business do
  use Oban.Worker, queue: :events

  @impl Oban.Worker
  def perform(%_{args: %{"id" => id}}) do
    model = MyApp.Repo.get(MyApp.Business.Man, id)

    case args do
      %{"in_the" => "business"} ->
        IO.inspect(model)

      %{"vote_for" => vote} ->
        IO.inspect([vote, model])

      _ ->
        IO.inspect(model)
    end

    :ok
  end
end
The use macro also accepts options to customize max attempts, priority, tags,
and uniqueness:
defmodule MyApp.LazyBusiness do
  use Oban.Worker,
    queue: :events,
    priority: 3,
    max_attempts: 3,
    tags: ["business"],
    unique: [period: 30]

  @impl Oban.Worker
  def perform(_job) do
    # do business slowly

    :ok
  end
end
Successful jobs should return :ok or an {:ok, value} tuple. The value
returned from perform/1 is used to control whether the job is treated as a
success, a failure, discarded completely or deferred until later.
See the Oban.Worker docs for more details on failure conditions and
Oban.Telemetry for details on job reporting.
Enqueueing Jobs
Jobs are simply Ecto structs and are enqueued by inserting them into the
database. For convenience and consistency all workers provide a new/2
function that converts an args map into a job changeset suitable for insertion:
%{id: 1, in_the: "business", of_doing: "business"}
|> MyApp.Business.new()
|> Oban.insert()
The worker's defaults may be overridden by passing options:
%{id: 1, vote_for: "none of the above"}
|> MyApp.Business.new(queue: :special, max_attempts: 5)
|> Oban.insert()
Jobs may be scheduled at a specific datetime in the future:
%{id: 1}
|> MyApp.Business.new(scheduled_at: ~U[2020-12-25 19:00:56.0Z])
|> Oban.insert()
Jobs may also be scheduled down to the second any time in the future:
%{id: 1}
|> MyApp.Business.new(schedule_in: 5)
|> Oban.insert()
Unique jobs can be configured in the worker, or when the job is built:
%{email: "brewster@example.com"}
|> MyApp.Mailer.new(unique: [period: 300, fields: [:queue, :worker])
|> Oban.insert()
Job priority can be specified using an integer from 0 to 3, with 0 being the
default and highest priority:
%{id: 1}
|> MyApp.Backfiller.new(priority: 2)
|> Oban.insert()
Any number of tags can be added to a job dynamically, at the time it is
inserted:
id = 1

%{id: id}
|> MyApp.OnboardMailer.new(tags: ["mailer", "record-#{id}"])
|> Oban.insert()
Multiple jobs can be inserted in a single transaction:
Ecto.Multi.new()
|> Oban.insert(:b_job, MyApp.Business.new(%{id: 1}))
|> Oban.insert(:m_job, MyApp.Mailer.new(%{email: "brewser@example.com"}))
|> Repo.transaction()
Occasionally you may need to insert a job for a worker that exists in another
application. In that case you can use Oban.Job.new/2 to build the changeset
manually:
%{id: 1, user_id: 2}
|> Oban.Job.new(queue: :default, worker: OtherApp.Worker)
|> Oban.insert()
Oban.insert/2,4 is the preferred way of inserting jobs as it provides some of
Oban's advanced features (i.e., unique jobs). However, you can use your
application's Repo.insert/2 function if necessary.
See Oban.Job.new/2 for a full list of job options.
Pruning Historic Jobs
Job stats and queue introspection are built on keeping job rows in the database
after they have completed. This allows administrators to review completed jobs
and build informative aggregates, at the expense of storage and an unbounded
table size. To prevent the oban_jobs table from growing indefinitely, Oban
provides active pruning of completed and discarded jobs.
By default, pruning retains jobs for 60 seconds.
Caveats & Guidelines
	Pruning is best-effort and performed out-of-band. This means that all limits
are soft; jobs beyond a specified age may not be pruned immediately after jobs
complete.

	Pruning is only applied to jobs that are completed or discarded (has
reached the maximum number of retries or has been manually killed). It'll
never delete a new job, a scheduled job or a job that failed and will be
retried.


Unique Jobs
The unique jobs feature lets you specify constraints to prevent enqueuing
duplicate jobs.  Uniqueness is based on a combination of args, queue,
worker, state and insertion time. It is configured at the worker or job
level using the following options:
	:period — The number of seconds until a job is no longer considered
duplicate. You should always specify a period. :infinity can be used to
indicate the job be considered a duplicate as long as jobs are retained.

	:fields — The fields to compare when evaluating uniqueness. The available
fields are :args, :queue and :worker, by default all three are used.

	:states — The job states that are checked for duplicates. The available
states are :available, :scheduled, :executing, :retryable and
:completed. By default all states are checked, which prevents any
duplicates, even if the previous job has been completed.


For example, configure a worker to be unique across all fields and states for 60
seconds:
use Oban.Worker, unique: [period: 60]
Configure the worker to be unique only by :worker and :queue:
use Oban.Worker, unique: [fields: [:queue, :worker], period: 60]
Or, configure a worker to be unique until it has executed:
use Oban.Worker, unique: [period: 300, states: [:available, :scheduled, :executing]]
Strong Guarantees
Unique jobs are guaranteed through transactional locks and database queries:
they do not rely on unique constraints in the database. This makes uniqueness
entirely configurable by application code, without the need for database
migrations.
Performance Note
If your application makes heavy use of unique jobs you may want to add an index
on the args column of the oban_jobs table. The other columns considered for
uniqueness are already covered by indexes.
Periodic Jobs
Oban allows jobs to be registered with a cron-like schedule and enqueued
automatically. Periodic jobs are registered as a list of {cron, worker} or
{cron, worker, options} tuples:
config :my_app, Oban, repo: MyApp.Repo, crontab: [
  {"* * * * *", MyApp.MinuteWorker},
  {"0 * * * *", MyApp.HourlyWorker, args: %{custom: "arg"}},
  {"0 0 * * *", MyApp.DailyWorker, max_attempts: 1},
  {"0 12 * * MON", MyApp.MondayWorker, queue: :scheduled, tags: ["mondays"]},
  {"@daily", MyApp.AnotherDailyWorker}
]
These jobs would be executed as follows:
	MyApp.MinuteWorker — Executed once every minute
	MyApp.HourlyWorker — Executed at the first minute of every hour with custom args
	MyApp.DailyWorker — Executed at midnight every day with no retries
	MyApp.MondayWorker — Executed at noon every Monday in the "scheduled" queue

The crontab format respects all standard rules and has one minute
resolution. Jobs are considered unique for most of each minute, which prevents
duplicate jobs with multiple nodes and across node restarts.
Cron Expressions
Standard Cron expressions are composed of rules specifying the minutes, hours,
days, months and weekdays. Rules for each field are comprised of literal values,
wildcards, step values or ranges:
	* — Wildcard, matches any value (0, 1, 2, ...)
	0 — Literal, matches only itself (only 0)
	*/15 — Step, matches any value that is a multiple (0, 15, 30, 45)
	0-5 — Range, matches any value within the range (0, 1, 2, 3, 4, 5)
	0-23/2 - Step values can be used in conjunction with ranges (0, 2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22)

Each part may have multiple rules, where rules are separated by a comma. The
allowed values for each field are as follows:
	minute — 0-59
	hour — 0-23
	days — 1-31
	month — 1-12 (or aliases, JAN, FEB, MAR, etc.)
	weekdays — 0-6 (or aliases, SUN, MON, TUE, etc.)

The following Cron extensions are supported:
	@hourly — 0 * * * *
	@daily (as well as @midnight) — 0 0 * * *
	@weekly — 0 0 * * 0
	@monthly — 0 0 1 * *
	@yearly (as well as @annually) — 0 0 1 1 *

Some specific examples that demonstrate the full range of expressions:
	0 * * * * — The first minute of every hour
	*/15 9-17 * * * — Every fifteen minutes during standard business hours
	0 0 * DEC * — Once a day at midnight during december
	0 7-9,4-6 13 * FRI — Once an hour during both rush hours on Friday the 13th

For more in depth information see the man documentation for cron and crontab
in your system.  Alternatively you can experiment with various expressions
online at Crontab Guru.
Caveats & Guidelines
	All schedules are evaluated as UTC unless a different timezone is configured.
See Oban.start_link/1 for information about configuring a timezone.

	Workers can be used for regular and scheduled jobs so long as they accept
different arguments.

	Duplicate jobs are prevented through transactional locks and unique
constraints. Workers that are used for regular and scheduled jobs must not
specify unique options less than 60s.

	Long running jobs may execute simultaneously if the scheduling interval is
shorter than it takes to execute the job. You can prevent overlap by passing
custom unique opts in the crontab config:


  custom_args = %{scheduled: true}

  unique_opts = [
    period: 60 * 60 * 24,
    states: [:available, :scheduled, :executing]
  ]

  config :my_app, Oban, repo: MyApp.Repo, crontab: [
    {"* * * * *", MyApp.SlowWorker, args: custom_args, unique: unique_opts},
  ]
Prioritizing Jobs
Normally, all available jobs within a queue are executed in the order they were
scheduled. You can override the normal behavior and prioritize or de-prioritize
a job by assigning a numerical priority.
	Priorities from 0-3 are allowed, where 0 is the highest priority and 3 is the
lowest.

	The default priority is 0, unless specified all jobs have an equally high
priority.

	All jobs with a higher priority will execute before any jobs with a lower
priority. Within a particular priority jobs are executed in their scheduled
order.


Testing
Oban provides some helpers to facilitate testing. The helpers handle the
boilerplate of making assertions on which jobs are enqueued. To use the
perform_job/2,3, assert_enqueued/1 and refute_enqueued/1 helpers in your
tests you must include them in your testing module and specify your app's Ecto
repo:
use Oban.Testing, repo: MyApp.Repo
Now you can assert, refute or list jobs that have been enqueued within your
integration tests:
assert_enqueued worker: MyWorker, args: %{id: 1}

# or

refute_enqueued queue: :special, args: %{id: 2}

# or

assert [%{args: %{"id" => 1}}] = all_enqueued worker: MyWorker
You can also easily unit test workers with the perform_job/2,3 function, which
automates validating job args, options, and worker results from a single
function call:
assert :ok = perform_job(MyWorker, %{id: 1})

# or

assert :ok = perform_job(MyWorker, %{id: 1}, attempt: 3, max_attempts: 3)

# or

assert {:error, _} = perform_job(MyWorker, %{bad: :arg})
See the Oban.Testing module for more details.
Caveats & Guidelines
As noted in Usage, there are some guidelines for running tests:
	Disable all job dispatching by setting queues: false or queues: nil in
your test.exs config. Keyword configuration is deep merged, so setting
queues: [] won't have any effect.

	Disable plugins via plugins: false. Default plugins, such as the fixed
pruner, aren't necessary in testing mode because jobs created within the
sandbox are rolled back at the end of the test. Additionally, the periodic
pruning queries will raise DBConnection.OwnershipError when the application
boots.

	Disable cron jobs via crontab: false. Periodic jobs aren't useful while
testing and scheduling can lead to random ownership issues.

	Be sure to use the Ecto Sandbox for testing. Oban makes use of database pubsub
events to dispatch jobs, but pubsub events never fire within a transaction.
Since sandbox tests run within a transaction no events will fire and jobs
won't be dispatched.


  config :my_app, MyApp.Repo, pool: Ecto.Adapters.SQL.Sandbox
Integration Testing
During integration testing it may be necessary to run jobs because they do work
essential for the test to complete, i.e. sending an email, processing media,
etc. You can execute all available jobs in a particular queue by calling
Oban.drain_queue/1 directly from your tests.
For example, to process all pending jobs in the "mailer" queue while testing
some business logic:
defmodule MyApp.BusinessTest do
  use MyApp.DataCase, async: true

  alias MyApp.{Business, Worker}

  test "we stay in the business of doing business" do
    :ok = Business.schedule_a_meeting(%{email: "monty@brewster.com"})

    assert %{success: 1, failure: 0} == Oban.drain_queue(:mailer)

    # Now, make an assertion about the email delivery
  end
end
See Oban.drain_queue/1 for additional details.
Error Handling
When a job returns an error value, raises an error or exits during execution the
details are recorded within the errors array on the job. When the number of
execution attempts is below the configured max_attempts limit, the job will
automatically be retried in the future.
The retry delay has an exponential backoff, meaning the job's second attempt
will be after 16s, third after 31s, fourth after 1m 36s, etc.
See the Oban.Worker documentation on "Customizing Backoff" for alternative
backoff strategies.
Error Details
Execution errors are stored as a formatted exception along with metadata about
when the failure occurred and which attempt caused it. Each error is stored with
the following keys:
	at The utc timestamp when the error occurred at
	attempt The attempt number when the error occurred
	error A formatted error message and stacktrace

See the Instrumentation docs for an example of
integrating with external error reporting systems.
Limiting Retries
By default, jobs are retried up to 20 times. The number of retries is controlled
by the max_attempts value, which can be set at the Worker or Job level. For
example, to instruct a worker to discard jobs after three failures:
use Oban.Worker, queue: :limited, max_attempts: 3
Limiting Execution Time
By default, individual jobs may execute indefinitely. If this is undesirable you
may define a timeout in milliseconds with the timeout/1 callback on your
worker module.
For example, to limit a worker's execution time to 30 seconds:
def MyApp.Worker do
  use Oban.Worker

  @impl Oban.Worker
  def perform(_args, _job) do
    something_that_may_take_a_long_time()

    :ok
  end

  @impl Oban.Worker
  def timeout(_job), do: :timer.seconds(30)
end
The timeout/1 function accepts an Oban.Job struct, so you can customize the
timeout using any job attributes.
Define the timeout value through job args:
def timeout(%_{args: %{"timeout" => timeout}}), do: timeout
Define the timeout based on the number of attempts:
def timeout(%_{attempt: attempt}), do: attempt * :timer.seconds(5)
Instrumentation and Logging
Oban provides integration with Telemetry, a dispatching library for
metrics. It is easy to report Oban metrics to any backend by attaching to
:oban events.
Here is an example of a sample unstructured log handler:
defmodule MyApp.ObanLogger do
  require Logger

  def handle_event([:oban, :job, :start], measure, meta, _) do
    Logger.warn("[Oban] :started #{meta.worker} at #{measure.system_time}")
  end

  def handle_event([:oban, :job, event], measure, meta, _) do
    Logger.warn("[Oban] #{event} #{meta.worker} ran in #{measure.duration}")
  end
end
Attach the handler to success and failure events in application.ex:
events = [[:oban, :job, :start], [:oban, :job, :stop], [:oban, :job, :exception]]

:telemetry.attach_many("oban-logger", events, &MyApp.ObanLogger.handle_event/4, [])
The Oban.Telemetry module provides a robust structured logger that handles all
of Oban's telemetry events. As in the example above, attach it within your
application.ex module:
:ok = Oban.Telemetry.attach_default_logger()
For more details on the default structured logger and information on event
metadata see docs for the Oban.Telemetry module.
Reporting Errors
Another great use of execution data is error reporting. Here is an example of
integrating with Honeybadger to report job failures:
defmodule ErrorReporter do
  def handle_event([:oban, :failure], measure, meta, _) do
    context =
      meta
      |> Map.take([:id, :args, :queue, :worker])
      |> Map.merge(measure)

    Honeybadger.notify(meta.error, context, meta.stack)
  end

  def handle_event([:oban, :trip_circuit], _measure, meta, _) do
    context = Map.take(meta, [:name])

    Honeybadger.notify(meta.error, context, meta.stack)
  end
end

:telemetry.attach_many(
  "oban-errors",
  [[:oban, :failure], [:oban, :trip_circuit]],
  &ErrorReporter.handle_event/4,
  nil
)
Isolation
Oban supports namespacing through PostgreSQL schemas, also called "prefixes" in
Ecto. With prefixes your jobs table can reside outside of your primary schema
(usually public) and you can have multiple separate job tables.
To use a prefix you first have to specify it within your migration:
defmodule MyApp.Repo.Migrations.AddPrefixedObanJobsTable do
  use Ecto.Migration

  def up do
    Oban.Migrations.up(prefix: "private")
  end

  def down do
    Oban.Migrations.down(prefix: "private")
  end
end
The migration will create the "private" schema and all tables, functions and
triggers within that schema. With the database migrated you'll then specify the
prefix in your configuration:
config :my_app, Oban,
  prefix: "private",
  repo: MyApp.Repo,
  queues: [default: 10]
Now all jobs are inserted and executed using the private.oban_jobs table. Note
that Oban.insert/2,4 will write jobs in the private.oban_jobs table, you'll
need to specify a prefix manually if you insert jobs directly through a repo.
Supervisor Isolation
Not only is the oban_jobs table isolated within the schema, but all
notification events are also isolated. That means that insert/update events will
only dispatch new jobs for their prefix. You can run multiple Oban instances
with different prefixes on the same system and have them entirely isolated,
provided you give each supervisor a distinct id.
Here we configure our application to start three Oban supervisors using the
"public", "special" and "private" prefixes, respectively:
def start(_type, _args) do
  children = [
    Repo,
    Endpoint,
    Supervisor.child_spec({Oban, name: ObanA, repo: Repo}, id: ObanA),
    Supervisor.child_spec({Oban, name: ObanB, repo: Repo, prefix: "special"}, id: ObanB),
    Supervisor.child_spec({Oban, name: ObanC, repo: Repo, prefix: "private"}, id: ObanC)
  ]

  Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
end
Community
There are a few places to connect and communicate with other Oban users.
	Request an invitation and join the #oban channel on Slack
	Ask questions and discuss Oban on the Elixir Forum
	Learn about bug reports and upcoming features in the issue tracker
	Follow @sorentwo on Twitter

Contributing
To run the Oban test suite you must have PostgreSQL 10+ running locally with a
database named oban_test. Follow these steps to create the database, create
the database and run all migrations:
mix test.setup
To ensure a commit passes CI you should run mix ci locally, which executes the
following commands:
	Check formatting (mix format --check-formatted)
	Lint with Credo (mix credo --strict)
	Run all tests (mix test --raise)
	Run Dialyzer (mix dialyzer)

  
Changelog
All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog
and this project adheres to Semantic Versioning.
Unreleased
2.0.0-rc.1 — 2020-06-12
Breaking Changes
	[Oban.Config] The :verbose setting is renamed to :log. The setting started
off as a simple boolean, but it has morphed to align with the log values
accepted by calls to Ecto.Repo.
To migrate, replace any :verbose declarations:
  config :my_app, Oban,
    verbose: false,
    ...
With use of :log instead:
  config :my_app, Oban,
    log: false,
    ...

	[Oban] The interface for start_queue/3 is replaced with start_queue/2 and
stop_queue/2 no longer accepts a queue name as the second argument. Instead,
both functions now accept a keyword list of options. This enables the new
local_only flag, which allows you to dynamically start and stop queues only
for the local node.
Where you previously called start_queue/2,3 or stop_queue/2 like this:
  :ok = Oban.start_queue(:myqueue, 10)
  :ok = Oban.stop_queue(:myqueue)
You'll now them with options, like this:
  :ok = Oban.start_queue(queue: :myqueue, limit: 10)
  :ok = Oban.stop_queue(queue: :myqueue)
Or, to only control the queue locally:
  :ok = Oban.start_queue(queue: :myqueue, limit: 10, local_only: true)
  :ok = Oban.stop_queue(queue: :myqueue, local_only: true)


2.0.0-rc.0 — 2020-06-03
Breaking Changes
	[Oban.Worker] The perform/2 callback is replaced with perform/1, where the
only argument is an Oban.Job struct. This unifies the interface for all
Oban.Worker callbacks and helps to eliminate confusion around pattern
matching on arguments.
To migrate change all worker definitions from accepting an args map and a
job struct:
  def perform(%{"id" => id}, _job), do: IO.inspect(id)
To accept a single job struct and match on the args key directly:
  def perform(%Job{args: %{"id" => id}}), do: IO.inspect(id)

	[Oban.Worker] The backoff/1 callback now expects a job struct instead of an
integer. That allows applications to finely control backoff based on more than
just the current attempt number. Use of backoff/1 with an integer is
no longer supported.
To migrate change any worker definitions that used a raw attempt like this:
  def backoff(attempt), do: attempt * 60
To match on a job struct instead, like this:
  def backoff(%Job{attempt: attempt}), do: attempt * 60

	[Oban.Telemetry] The format for telemetry events has changed to match the new
telemetry span convention. This listing maps the old event to the new one:
	[:oban, :started] -> [:oban, :job, :start]
	[:oban, :success] -> [:oban, :job, :stop]
	[:oban, :failure] -> [:oban, :job, :exception]
	[:oban, :trip_circuit] -> [:oban, :circuit, :trip]
	[:oban, :open_circuit] -> [:oban, :circuit, :open]

In addition, for exceptions the stacktrace meta key has changed from :stack
to the standardized :stacktrace.

	[Oban.Prune] Configurable pruning is no longer available. Instead, pruning is
handled by the new plugin system. A fixed period pruning module is enabled as
a default plugin. Th plugin always retains "prunable" (discarded or complete)
jobs for 60 seconds.
Remove any :prune, :prune_interval or prune_limit settings from your
config. To disable the pruning plugin in test mode set plugins: false
instead.

	[Oban.Beat] Pulse tracking and periodic job rescue are no longer available.
Pulse tracking and rescuing will be handled by an external plugin. This is
primarily an implementation detail, but it means that jobs may be left in the
executing state after a crash or forced shutdown.
Remove any :beats_maxage, :rescue_after or :rescue_interval settings
from your config.


Fixed
	[Oban.Scheduler] Ensure isolation between transaction locks in different
prefixes. A node with multiple prefix-isolated instances (i.e. "public" and
"private") would always attempt to schedule cron jobs at the same moment. The
first scheduler would acquire a lock and block out the second, preventing the
second scheduler from ever scheduling jobs.

	[Oban.Query] Correctly prefix unprepared unique queries. Unique queries always
targeted the "public" prefix, which either caused incorrect results when there
were both "public" and an alternate prefix. In situations where there wasn't
a public oban_jobs table at all it would cause cryptic transaction errors.

	[Oban.Query] Wrap all job fetching in an explicit transaction to enforce FOR UPDATE SKIP LOCKED semantics. Prior to this it was possible to run the same
job at the same time on multiple nodes.

	[Oban.Crontab] Fix weekday matching for Sunday, which is represented as 0 in
crontabs.


Added
	[Oban] Bubble up errors and exits when draining queues by passing
with_safety: false as an option to Oban.drain_queue/3.

	[Oban] Add Oban.cancel_job/2 for safely discarding scheduled jobs or killing
executing jobs. This deprecates Oban.kill_job/2, which isn't as flexible.

	[Oban.Worker] Support returning {:snooze, seconds} from perform/2 to
re-schedule a job some number of seconds in the future. This is useful for
recycling jobs that aren't ready to run yet, e.g. because of rate limiting.

	[Oban.Worker] Support returning :discard from perform/1 to immediately
discard a job. This is useful when a job encounters an error that won't
resolve with time, e.g. invalid arguments or a missing record.

	[Oban.Job] Introduce a virtual unsaved_error field, which is populated with
an error map after failed execution. The unsaved_error field is set before
any calls to the worker's backoff/1 callback, allowing workers to calculate
a custom backoff depending on the error that failed the job.

	[Oban.Worker] Add :infinity option for unique period.

	[Oban.Telemetry] Add span/3 for reporting normalized :start, :stop and
:exception events with timing information.

	[Oban.Telemetry] Include the configured prefix in all event metadata. This
makes it possible to identify which schema prefix a job ran with, which is
useful for differentiating errors in a multi-tenant system.

	[Oban.Telemetry] Include queue_time as a measurement with stop and
exception events. This is a measurement in milliseconds of the amount of
time between when a job was scheduled to run and when it was last attempted.

	[Oban.Testing] Add perform_job/2,3 helper to automate validating,
normalizing and performing jobs while unit testing. This is now the preferred
way to unit test workers.
To update your tests replace any calls to perform/1,2 with the new
Oban.Testing.perform_job/2,3 helper:


  defmodule MyApp.WorkerTest do
    use MyApp.DataCase, async: true

    use Oban.Testing, repo: MyApp.Repo

    alias MyApp.Worker

    test "doing business in my worker" do
      assert :ok = perform_job(Worker, %{id: 1})
    end
  end
  The perform_job/2,3 helper will verify the worker, the arguments and any
  provided options. It will then verify that your worker returns a valid result
  and return the value for you to assert on.
	[Oban.Crontab] Add support for non-standard expressions such as @daily,
@hourly, @midnight, etc.

	[Oban.Crontab] Add support for using step values in conjunction with ranges,
enabling expressions like 10-30/2, 15-45/3, etc.


Changed
	[Oban.Notifier] Make the module public and clean up the primary function
interfaces. Listening for and delivering notifications is simplified and no
longer requires macros for pattern matching.
Notifier dispatching performance is slightly improved as well. It is now a
no-op if no processes are listening to a notification's channel.

	[Oban.Query] The completed_at timestamp is no longer set for failed jobs,
whether they are put in the discarded or retryable state. However, the
information is still available and is recorded in the errors array as the
at value with the error for that attempt.
This corrects a long standing inconsistency between discarding a job manually
or automatically when it exhausts retries.

	[Oban.Producer] Stop dispatching jobs immediately on queue startup. Instead,
only dispatch on the first poll. This makes it possible to send the producer
a message or allow sandboxed connection access before the initial dispatch.

	[Oban.Worker] Limit default backoff calculations to 20 attempts, or roughly 24
days. The change addresses an issue with snoozing, which can increase a job's
attempts into the hundreds or thousands. In this situation the algorithm
calculates the backoff using a ratio of attempts to max attempts, but is still
limited to roughly 24 days.


1.2.0 — 2020-03-05
Fixed
	[Oban] Handle the :shutdown message when a job is killed purposefully.
Previously the message was ignored, which caused the producer to keep a
reference to the discarded job and prevented dispatching additional jobs.

Added
	[Oban.Testing] Add assert_enqueued/2 and refute_enqueued/2 to allow
asserting with a timeout, like assert_received.

	[Oban.Telemetry] Add [:oban, :started] event to report the system time as
start_time when jobs start. This enables span tracking for jobs, improving
integration with monitoring tools like NewRelic.


1.1.0 — 2020-02-17
Fixed
	[Oban.Crontab] Allow any number of spaces and/or tabs in cron expressions.

	[Oban.Pruner] Prevent deadlocks while pruning by ensuring that only a single
node can prune at any given time.

	[Oban.Queue.Executor] Improve handling of nested/linked process failures.
Previously if a job spawned a process that crashed (i.e. through Task.async)
then the job was left stuck in an executing state.


Added
	[Oban.Worker] Add timeout/1 callback for limiting how long a job may
execute. The default value is :infinity, which allows a job to run
indefinitely and mirrors the previous behavior.

	[Oban.Telemetry] Add :error and :stack to trip_circuit event metadata.

	[Oban.Queue.Executor] Retry success/failure database updates after a job
finishes.
On occasion something may happen to the database connection and updating a
job's state would fail. Now we retry writing with a linear backoff to prevent
the job from getting stuck in an executing state.


Changed
	[Oban.Worker] Tighten the spec for perform/2. Now workers are expected to
return only :ok, {:ok, result} or {:error, reason}. A warning is logged
if any other value is returned—for high throughput systems this may cause
performance issues and you should update your worker's return values.
Returning a success tuple is supported for testing purposes and backward
compatibility, while returning :ok on success if preferred.


1.0.0 — 2020-01-29
No changes from 1.0.0-rc.2.
1.0.0-rc.2 — 2020-01-22
Fixed
	[Oban.Migration] Separate adding and modifying new columns in the V8
migration. The columns can't be modified without a flush.

Changed
	[Oban.Testing] Accept a custom prefix when making test assertions.

1.0.0-rc.1 — 2020-01-21
Migration Required (V8)
This is the first required migration since 0.8.0, released in 09/2019. It brings
with it a new column, discarded_at, a streamlined notifications trigger, job
prioritiy and job tags.
Upgrading only requires running the new migration.
First, generate a new migration:
mix ecto.gen.migration upgrade_oban_jobs_to_v8
Next, call Oban.Migrations in the generated migration:
defmodule MyApp.Repo.Migrations.UpdateObanJobsToV8 do
  use Ecto.Migration

  def up, do: Oban.Migrations.up(version: 8)
  def down, do: Oban.Migrations.down()
end
Oban will manage upgrading to V8 regardless of the version your application is
currently using, and it will roll back a single version.
Added
	[Oban] Add timezone support for scheduling cronjobs using timezones other
than "Etc/UTC". Using a custom timezone requires a timezone database such as
tzdata.

	[Oban] Add dispatch_cooldown option to configure the minimum time between
a producer fetching more jobs to execute.

	[Oban] Add beats_maxage option to configure how long heartbeat rows are
retained in the oban_beats table. Each queue generates one row per second,
so rows may accumulate quickly. The default value is now five minutes, down
from one hour previously.

	[Oban.Job] Add discarded_at timestamp to help identify jobs that were
discarded and not completed. The timestamp is added by the V8 migration and it
is also included in the original create table from V1 as a minor space
saving optimization (packing datetime columns together because they use a
predictable 4bytes of space).

	[Oban.Job] Add numerical priority value to control the order of execution
for jobs within a queue. The priority can be set between 0 and 3, with 0
being the default and the highest priority.

	[Oban.Job] Add tags field for arbitrarily organizing associated tags. Tags
are a list of strings stored as an array in the database, making them easy
to search and filter by.


Changed
	[Oban] Change the default prune value from :disabled to {:maxlen, 1_000}. Many people don't change the default until they realize that a lot of
jobs are lingering in the database. It is rare that anybody would want to keep
all of their jobs forever, so a conservative default is better than
:disabled.

	[Oban] Change oban_beats retention from one hour to five minutes. The value
is now configurable, with a default of 300s. The lower bound is 60s
because we require one minute of heartbeat activity to rescue orphaned jobs.

	[Oban.Queue.Producer] Introduce "dispatch cooldown" as a way to debounce
repeatedly fetching new jobs. Repeated fetching floods the producer's message
queue and forces the producer to repeatedly fetch one job at a time, which is
not especially efficient. Debounced fetching is much more efficient for the
producer and the database, increasing maximum jobs/sec throughput so that it
scales linearly with a queue's concurrency settings (up to what the database
can handle).

	[Oban.Query] Discard jobs that have exhausted the maximum attempts rather than
rescuing them. This prevents repeatedly attempting a job that has consistently
crashed the BEAM.

	[Oban.Query] Use transactional locks to prevent duplicate inserts without
relying on unique constraints in the database. This provides strong unique
guarantees without requiring migrations.


Removed
	[Oban.Notifications] An overhauled and simplified insert trigger no longer
emits update notifications. This was largely an internal implementation
detail and wasn't publicly documented, but it does effect the UI.

0.12.1 — 2019-12-13
Fixed
	[Oban.Worker] Deep merge unique options between the worker and custom
params. Previously the unique options passed to Worker.new/2 would
completely override the options stored by use Oban.Worker. This was
primarily an issue for crontab jobs, where the period is always passed by the
scheduler.

Changed
	[Oban] Allow setting crontab: false or crontab: nil to disable scheduling.
This matches the queues behavior and makes it possible to override the
default configuration within each environment, i.e. when testing.

	[Oban.Testing] Better error message for Oban.Testing.assert_enqueued/2


0.12.0 – 2019-11-26
Migration Optional (V7)
The queries used to prune by limit and age are written to utilize a single
partial index for a huge performance boost on large tables. The new V7 migration
will create the index for you—but that may not be ideal for tables with millions
of completed or discarded jobs because it can't be done concurrently.
If you have an extremely large jobs table you can add the index concurrently in
a dedicated migration:
create index(
         :oban_jobs,
         ["attempted_at desc", :id],
         where: "state in ('completed', 'discarded')",
         name: :oban_jobs_attempted_at_id_index,
         concurrently: true
       )
Added
	[Oban] Add start_queue/3 and stop_queue/2 for dynamically starting and
stopping supervised queues across nodes.

	[Oban] Add drain_queue/3 to accept drain options. with_scheduled: true
allows draining scheduled jobs.

	[Oban] Expose circuit_backoff as a "twiddly" option that controls how long
tripped circuit breakers wait until re-opening.

	[Oban.Testing] Accept a value/delta tuple for testing timestamp fields. This
allows more robust testing of timestamps such as scheduled_at.

	[Oban.Telemetry] Emit [:oban, :trip_circuit] and [:oban, :open_circuit]
events for circuit breaker activity. Previously an error was logged when the
circuit was tripped, but there wasn't any way to monitor circuit breakers.
Circuit breaker activity is logged by the default telemetry logger (both
:trip_circuit and :open_circuit events).


Fixed
	[Oban.Query] Avoid using prepared statements for all unique queries. This
forces Postgres to use a custom plan (which utilizes the compound index)
rather than falling back to a generic plan.

	[Oban.Job] Include all permitted fields when converting a Job to a map,
preserving any optional values that were either specified by the user or came
via Worker defaults.

	[Oban.Migrations] Guard against missing migration modules in federated
environments.


Changed
	[Oban] Allow the multi name provided to Oban.insert/3,4 to be any term,
not just an atom.

	[Oban.Query] Use a consistent and more performant set of queries for pruning.
Both pruning methods are optimized to utilize a single partial index.


0.11.1 — 2019-11-13
Fixed
	[Oban.Pruner] Apply prune_limit when pruning beats. A lot of beats can
accumulate when pruning has been disabled. Later, when pruning is enabled,
the pruner would try to delete too many beats and would time out.

	[Oban.Crontab.Scheduler] Use a zero arity function for the transaction
callback. The one arity version is only available in Ecto >= 3.2.


0.11.0 — 2019-11-06
Migration Optional (V6)
Job id's greater than 2,147,483,647 (PG int limit) can't be inserted into the
running array on oban_beats. The array that Ecto defines uses int instead of
bigint, which can't store the larger integers. This migration changes the
column type to bigint[], a locking operation that may take a few seconds.
Added
	[Oban] Added crontab support for automatically enqueuing jobs on a fixed
schedule. A combination of transactional locks and unique jobs prevents
scheduling duplicate jobs.

Fixed
	[Oban.Migrations] Add a comment when migrating oban_jobs to V5 and when
rolling back down to V4.

	[Oban.Query] Apply the configured log level to unique queries.

	[Oban.Notifier] Prevent open connections from accumulating when the circuit is
tripped during the connection phase. This change may leave notifications in a
state where they aren't listening to all channels.


Changed
	[Oban.Notifier] Replay oban_update notifications to subscribed processes.

0.10.1 — 2019-10-08
Changed
	[Oban.Notifier] Replay oban_gossip notifications to subscribed processes.

0.10.0 — 2019-10-03
Migration Optional (V5)
Tables with a lot of available jobs (hundreds of thousands to several million)
are prone to time outs when fetching new jobs. The planner fails to optimize
using the index available on queue, state and scheduled_at, forcing both a
slow sort pass and an expensive bitmap heap scan.
This migration drops the separate indexes in favor of a a single composite
index. The resulting query is up to 258,757x faster on large tables while
still usable for all of the other maintenance queries.
History of the EXPLAIN ANALYZE output as the query was optimized is available
here: https://explain.depesz.com/s/9Vh7
Changed
	[Oban.Config] Change the default for verbose from true to false. Also,
:verbose now accepts only false and standard logger levels.  This change
aims to prevent crashes due to conflicting levels when the repo's log level is
set to false.

Fixed
	[Oban.Notifier] Restructure the notifier in order to to isolate producers from
connection failures. Errors or loss of connectivity in the notification
connection no longer kills the notifier and has no effect on the producers.

0.9.0 — 2019-09-20
Added
	[Oban] Add insert_all/2 and insert_all/4, corresponding to
Ecto.Repo.insert_all/3 and Ecto.Multi.insert_all/5, respectively.

	[Oban.Job] Add to_map/1 for converting a changeset into a map suitable for
database insertion. This is used by Oban.insert_all/2,4 internally and is
exposed for convenience.


Changed
	[Oban.Config] Remove the default queue value of [default: 10], which was
overriden by Oban.start_link/1 anyhow.

	[Oban.Telemetry] Allow the log level to be customized when attaching the
default logger. The default level is :info, the same as it was before.


Fixed
	[Oban.Migrations] Prevent invalid up and down targets when attempting to
run migrations that have already been ran. This was primarily an issue in CI,
where the initial migration was unscoped and would migrate to the current
version while a subsequent migration would attempt to migrate to a lower
version.

	[Oban.Job] Prevent a queue comparison with nil by retaining the default
queue (default) when building uniqueness checks.

	[Oban.Job] Set state to scheduled for jobs created with a scheduled_at
timestamp. Previously the state was only set when schedule_in was used.


0.8.1 — 2019-09-11
Changed
	[Oban.Notifier] Restore the gossip macro and allow oban_gossip channel for
notifications.

Fixed
	[Oban.Migrations] Prevent invalid up and down ranges when repeatedly
migrating without specifying a version. This issue was seen when running all
of the up migrations on a database from scratch, as there would be multiple
oban migrations that simply delegated to up and down.

0.8.0 — 2019-09-06
Migration Required (V4)
This release requires a migration to V3, with an optional migration to V4.
V3 adds a new column to jobs and creates the oban_beats table, while V4 drops
a function used by the old advisory locking system.
For a smooth zero-downtime upgrade, migrate to V3 first and then V4 in a separate
release. The following sample migration will only upgrade to V3:
defmodule MyApp.Repo.Migrations.AddObanBeatsTable do
  use Ecto.Migration

  def up do
    Oban.Migrations.up(version: 3)
  end

  def down do
    Oban.Migrations.down(version: 2)
  end
end
Added
	[Oban.Job] Add an attempted_by column used to track the node, queue and
producer nonce that attempted to execute a job.

	[Oban.Beat] Add a new oban_beats table, used by producers to publish "pulse"
information including the node, queue, running jobs and other information
previously published by gossip notifications.
Beat records older than one hour are pruned automatically. Beat pruning
respects the :disabled setting, but it ignores length and age configuration.
The goal is to prevent bloating the table with useless historic
information—each queue generates one beat a second, 3,600 beat records per
hour even when the queue is idle.


Changed
	[Oban.Executor] Don't swallow an ArgumentError when raised by a worker's
backoff function.

	[Oban.Notifier] Remove gossip notifications entirely, superseded by pulse
activity written to oban_beats.

	[Oban.Query] Remove all use of advisory locks!

	[Oban.Producer] Periodically attempt to rescue orphans, not just at startup.
By default a rescue is attempted once a minute and it checks for any executing
jobs belonging to a producer that hasn't had a pulse for more than a minute.


Fixed
	[Oban.Worker] Validate worker options after the module is compiled. This
allows dynamic configuration of compile time settings via module attributes,
functions, Application.get_env/3, etc.

	[Oban.Query] Remove scheduled_at check from job fetching query. This could
prevent available jobs from dispatching when the database's time differed from
the system time.

	[Oban.Migrations] Fix off-by-one error when detecting the version to migrate
up from.


0.7.1 — 2019-08-15
Fixed
	[Oban.Query] Release advisory locks in batches rather than individually after
a job finishes execution. By tracking unlockable jobs and repeatedly
attempting to unlock them for each connection we ensure that eventually all
advisory locks are released.
The previous unlocking system leaked advisory locks at a rate proportional to
the number of connections in the db pool. The more connections, the more locks
that wouldn't release. With a default value of 64 for max_locks_per_transaction
the database would raise "ERROR:  53200: out of shared memory" after it hit a
threshold (11,937 exactly, in my testing).

	[Oban.Worker] Allow max_attempts to be 1 or more. This used to be possible
and was broken unintentionally by compile time validations.


0.7.0 — 2019-08-08
Added
	[Oban] Added insert/2, insert!/2 and insert/4 as a convenient and more
powerful way to insert jobs. Features such as unique jobs and the upcoming
prefix support only work with insert.

	[Oban] Add prefix support. This allows entirely isolated job queues within
the same database.

	[Oban.Worker] Compile time validation of all passed options. Catch typos and
other invalid options when a worker is compiled rather than when a job is
inserted for the first time.

	[Oban.Worker] Unique job support through the unique option. Set a unique
period, and optionally fields and states, to enforce uniqueness within a
window of time. For example, to make a job unique by args, queue and worker
for 2 minutes:
use Oban.Worker, unique: [period: 120, fields: [:args, :queue, :worker]]
Note, unique support relies on the use of Oban.insert/2,4.


Changed
	[Oban.Worker] Remove the perform/1 callback in favor of perform/2. The new
perform/2 function receives the job's args, followed by the complete job
struct. This new function signature makes it clear that the args are always
available, and the job struct is also there when it is needed. A default
perform/2 function is not generated automatically by the use macro and
must be defined manually.
This is a breaking change and all worker modules will need to be updated.
Thankfully, due to the behaviour change, warnings will be emitted when you
compile after the upgrade.
If your perform functions weren't matching on the Oban.Job struct then
you can migrate your workers by adding a second _job argument:
def perform(%{"some" => args}, _job)
If you were making use of Oban.Job metadata in perform/1 then you can move
the job matching to the second argument:
def perform(_args, %{attempt: attempt})
See the issue that suggested this change for more details and
discussion.

	[Oban.Producer] Use send_after/3 instead of :timer.send_interval/2 to
maintain scheduled dispatch. This mechanism is more accurate under system
load and it prevents :poll messages from backing up for each producer.

	[Oban.Migration] Accept a keyword list with :prefix and :version as
options rather than a single version string. When a prefix is supplied the
migration will create all tables, indexes, functions and triggers within that
namespace. For example, to create the jobs table within a "private" prefix:
Oban.Migrate.up(prefix: "private")


0.6.0 — 2019-07-26
Added
	[Oban.Query] Added :verbose option to control general query logging. This
allows debug query activity within Oban to be silenced during testing and
development.

	[Oban.Testing] Added all_enqueued/1 helper for testing. The helper returns
a list of jobs matching the provided criteria. This makes it possible to test
using pattern matching, which is more flexible than a literal match within the
database.


Changed
	[Oban.Config] All passed options are validated. Any unknown options will raise
an ArgumentError and halt startup. This prevents misconfiguration through
typos and passing unsupported options.

	[Oban.Worker] The perform/1 function now receives an Oban.Job struct as
the sole argument, calling perform/1 again with only the args map if no
clause catches the struct. This allows workers to use any attribute of the job
to customize behaviour, e.g. the number of attempts or when a job was inserted
into the database.
The implementation is entirely backward compatible, provided workers are
defined with the use macro. Workers that implement the Oban.Worker
behaviour manually will need to change the signature of perform/1 to accept
a job struct.

	[Oban] Child process names are generated from the top level supervisor's name,
i.e. setting the name to "MyOban" on start_link/1 will set the notifier's
name to MyOban.Notifier. This improves isolation and allows multiple
supervisors to be ran on the same node.


Fixed
	[Oban.Producer] Remove duplicate polling timers. As part of a botched merge
conflict resolution two timers were started for each producer.

0.5.0 — 2019-06-27
Added
	[Oban.Pruner] Added :prune_limit option to constrain the number of rows
deleted on each prune iteration. This prevents locking the database when there
are a large number of jobs to delete.

Changed
	[Oban.Worker] Treat {:error, reason} tuples returned from perform/1 as a
failure. The :kind value reported in telemetry events is now differentiated,
where a rescued exception has the kind :exception, and an error tuple has
the kind :error.

Fixed
	[Oban.Testing] Only check available and scheduled jobs with the
assert|refute_enqueued testing helpers.

	[Oban.Queue.Watchman] Catch process exits when attempting graceful shutdown.
Any exits triggered within terminate/2 are caught and ignored. This safely
handles situations where the producer exits before the watchman does.

	[Oban.Queue.Producer] Use circuit breaker protection for gossip events and
call queries directly from the producer process. This prevents pushing all
queries through the Notifier, and ensures more granular control over gossip
errors.


0.4.0 — 2019-06-10
Added
	[Oban] Add Oban.drain_queue/1 to help with integration testing. Draining a
queue synchronously executes all available jobs in the queue from within the
calling process. This avoids any sandbox based database connection issues and
prevents race conditions from asynchronous processing.

	[Oban.Worker] Add backoff/1 callback and switch to exponential backoff with
a base value as the default. This allows custom backoff timing for individual
workers.

	[Oban.Telemetry] Added a new module to wrap a default handler for structured
JSON logging. The log handler is attached by calling
Oban.Telemetry.attach_default_logger/0 somewhere in your application code.

	[Oban.Queue.Producer] Guard against Postgrex errors in all producer queries
using a circuit breaker. Failing queries will no longer crash the producer.
Instead, the failure will be logged as an error and it will trip the
producer's circuit breaker. All subsequent queries will be skipped until the
breaker is enabled again approximately a minute later.
This feature simplifies the deployment process by allowing the application to
boot and stay up while Oban migrations are made. After migrations have
finished each queue producer will resume making queries.


Changed
	[Oban] Telemetry events now report timing as %{duration: duration} instead
of %{timing: timing}. This aligns with the telemetry standard of using
duration for the time to execute something.

	[Oban] Telemetry events are now broken into success and failure at the
event level, rather than being labeled in the metadata. The full event names
are now [:oban, :success] and [:oban, :failure].

	[Oban.Job] Rename scheduled_in to schedule_in for readability and
consistency. Both the Oban docs and README showed schedule_in, which reads
more clearly than scheduled_in.

	[Oban.Pruner] Pruning no longer happens immediately on startup and may be
configured through the :prune_interval option. The default prune interval is
still one minute.


Fixed
	[Oban.Migrations] Make partial migrations more resilient by guarding against
missing versions and using idempotent statements.

0.3.0 - 2019-05-29
Migration Required (V2)
Added
	[Oban] Allow setting queues: false or queues: nil to disable queue
dispatching altogether. This makes it possible to override the default
configuration within each environment, i.e. when testing.
The docs have been updated to promote this mechanism, as well as noting that
pruning must be disabled for testing.

	[Oban.Testing] The new testing module provides a set of helpers to make
asserting and refuting enqueued jobs within tests much easier.


Changed
	[Oban.Migrations] Explicitly set id as a bigserial to avoid mistakenly
generating a uuid primary key.

	[Oban.Migrations] Use versioned migrations that are immutable. As database
changes are required a new migration module is defined, but the interface of
Oban.Migrations.up/0 and Oban.Migrations.down/0 will be maintained.
From here on all releases with database changes will indicate that a new
migration is necessary in this CHANGELOG.

	[Oban.Query] Replace use of (bigint) with (int, int) for advisory locks.
The first int acts as a namespace and is derived from the unique oid value
for the oban_jobs table. The oid is unique within a database and even
changes on repeat table definitions.
This change aims to prevent lock collision with application level advisory
lock usage and other libraries. Now there is a 1 in 2,147,483,647 chance of
colliding with other locks.

	[Oban.Job] Automatically remove leading "Elixir." namespace from stringified
worker name. The prefix complicates full text searching and reduces the score
for trigram matches.


0.2.0 - 2019-05-15
Added
	[Oban] Add pause_queue/2, resume_queue/2 and scale_queue/3 for
dynamically controlling queues.
	[Oban] Add kill_job/2 for terminating running jobs and marking them as
discarded.
	[Oban] Add config/0 for retreiving the supervisors config. This is primarily
useful for integrating oban into external applications.
	[Oban.Queue.Producer] Use database triggers to immediately dispatch when a job
is inserted into the oban_jobs table.
	[Oban.Job] Execution errors are stored as a jsonb array for each job. Each
error is stored, not just the most recent one. Error entries contains these
keys:
	at The utc timestamp when the error occurred at
	attempt The attempt number when the error ocurred
	error A formatted error message and stacktrace, passed through
Exception.blame/3


	[Oban.Config] Validate all options based on type and allowed values. Any
invalid option will raise, preventing supervisor boot.
	[Oban.Notifier] Broadcast runtime gossip through pubsub, allowing any
external system to get stats at the node and queue level.

Changed
	[Oban.Queue.Supervisor] Set the min_demand to 1 for all consumers. This
ensures that each queue will run the configured number of concurrent jobs. By
default the min_demand is half of max_demand, which means a few slow jobs
can prevent the queue from running the expected number of concurrent jobs.
	[Oban.Job] Change psuedo-states based on job properties into fixed states,
this applies to scheduled and retryable.
	[Oban.Job] The "Elixir" prefix is stripped from worker names before storing
jobs in the database. Module lookup performs the same way, but it cleans up
displaying the worker name as a string.
	[Oban.Job] Accept all job fields as changeset parameters. While not
encouraged for regular use, this is essential for testing various states.

0.1.0 - 2019-03-10
	[Oban] Initial release with base functionality.

  
    
Oban    



      
Oban is a robust job processing library which uses PostgreSQL for storage and
coordination.
Each Oban instance is a supervision tree and not an application. That means it
won't be started automatically and must be included in your application's
supervision tree. All of your configuration is passed into the supervisor,
allowing you to configure Oban like the rest of your application:
# config/config.exs
config :my_app, Oban,
  repo: MyApp.Repo,
  queues: [default: 10, events: 50, media: 20]

# lib/my_app/application.ex
defmodule MyApp.Application do
  @moduledoc false

  use Application

  alias MyApp.Repo
  alias MyAppWeb.Endpoint

  def start(_type, _args) do
    children = [
      Repo,
      Endpoint,
      {Oban, oban_config()}
    ]

    Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
  end

  defp oban_config do
    opts = Application.get_env(:my_app, Oban)

    # Prevent running queues or scheduling jobs from an iex console, i.e. when starting app with `iex -S mix`
    if Code.ensure_loaded?(IEx) and IEx.started?() do
      opts
      |> Keyword.put(:crontab, false)
      |> Keyword.put(:queues, false)
    else
      opts
    end
  end
end
If you are running tests (which you should be) you'll want to disable pruning
, enqueuing scheduled jobs and job dispatching altogether when testing:
# config/test.exs
config :my_app, Oban, crontab: false, queues: false, plugins: false
Configuring Queues
Queues are specified as a keyword list where the key is the name of the queue
and the value is the maximum number of concurrent jobs. The following
configuration would start four queues with concurrency ranging from 5 to 50:
queues: [default: 10, mailers: 20, events: 50, media: 5]
There isn't a limit to the number of queues or how many jobs may execute
concurrently in each queue. Here are a few caveats and guidelines:
Caveats & Guidelines
	Each queue will run as many jobs as possible concurrently, up to the
configured limit. Make sure your system has enough resources (i.e. database
connections) to handle the concurrent load.

	Queue limits are local (per-node), not global (per-cluster). For example,
running a queue with a local limit of one on three separate nodes is
effectively a global limit of three. If you require a global limit you must
restrict the number of nodes running a particular queue.

	Only jobs in the configured queues will execute. Jobs in any other queue will
stay in the database untouched.

	Be careful how many concurrent jobs make expensive system calls (i.e. FFMpeg,
ImageMagick). The BEAM ensures that the system stays responsive under load,
but those guarantees don't apply when using ports or shelling out commands.


Defining Workers
Worker modules do the work of processing a job. At a minimum they must define a
perform/2 function, which is called with an args map and the job struct.
Note that the args map passed to perform/2 will always have string keys,
regardless of the key type when the job was enqueued. The args are stored as
jsonb in PostgreSQL and the serialization process automatically stringifies
all keys.
Define a worker to process jobs in the events queue:
defmodule MyApp.Business do
  use Oban.Worker, queue: :events

  @impl Oban.Worker
  def perform(%_{args: %{"id" => id}}) do
    model = MyApp.Repo.get(MyApp.Business.Man, id)

    case args do
      %{"in_the" => "business"} ->
        IO.inspect(model)

      %{"vote_for" => vote} ->
        IO.inspect([vote, model])

      _ ->
        IO.inspect(model)
    end

    :ok
  end
end
The use macro also accepts options to customize max attempts, priority, tags,
and uniqueness:
defmodule MyApp.LazyBusiness do
  use Oban.Worker,
    queue: :events,
    priority: 3,
    max_attempts: 3,
    tags: ["business"],
    unique: [period: 30]

  @impl Oban.Worker
  def perform(_job) do
    # do business slowly

    :ok
  end
end
Successful jobs should return :ok or an {:ok, value} tuple. The value
returned from perform/1 is used to control whether the job is treated as a
success, a failure, discarded completely or deferred until later.
See the Oban.Worker docs for more details on failure conditions and
Oban.Telemetry for details on job reporting.
Enqueueing Jobs
Jobs are simply Ecto structs and are enqueued by inserting them into the
database. For convenience and consistency all workers provide a new/2
function that converts an args map into a job changeset suitable for insertion:
%{id: 1, in_the: "business", of_doing: "business"}
|> MyApp.Business.new()
|> Oban.insert()
The worker's defaults may be overridden by passing options:
%{id: 1, vote_for: "none of the above"}
|> MyApp.Business.new(queue: :special, max_attempts: 5)
|> Oban.insert()
Jobs may be scheduled at a specific datetime in the future:
%{id: 1}
|> MyApp.Business.new(scheduled_at: ~U[2020-12-25 19:00:56.0Z])
|> Oban.insert()
Jobs may also be scheduled down to the second any time in the future:
%{id: 1}
|> MyApp.Business.new(schedule_in: 5)
|> Oban.insert()
Unique jobs can be configured in the worker, or when the job is built:
%{email: "brewster@example.com"}
|> MyApp.Mailer.new(unique: [period: 300, fields: [:queue, :worker])
|> Oban.insert()
Job priority can be specified using an integer from 0 to 3, with 0 being the
default and highest priority:
%{id: 1}
|> MyApp.Backfiller.new(priority: 2)
|> Oban.insert()
Any number of tags can be added to a job dynamically, at the time it is
inserted:
id = 1

%{id: id}
|> MyApp.OnboardMailer.new(tags: ["mailer", "record-#{id}"])
|> Oban.insert()
Multiple jobs can be inserted in a single transaction:
Ecto.Multi.new()
|> Oban.insert(:b_job, MyApp.Business.new(%{id: 1}))
|> Oban.insert(:m_job, MyApp.Mailer.new(%{email: "brewser@example.com"}))
|> Repo.transaction()
Occasionally you may need to insert a job for a worker that exists in another
application. In that case you can use Oban.Job.new/2 to build the changeset
manually:
%{id: 1, user_id: 2}
|> Oban.Job.new(queue: :default, worker: OtherApp.Worker)
|> Oban.insert()
Oban.insert/2,4 is the preferred way of inserting jobs as it provides some of
Oban's advanced features (i.e., unique jobs). However, you can use your
application's Repo.insert/2 function if necessary.
See Oban.Job.new/2 for a full list of job options.
Pruning Historic Jobs
Job stats and queue introspection are built on keeping job rows in the database
after they have completed. This allows administrators to review completed jobs
and build informative aggregates, at the expense of storage and an unbounded
table size. To prevent the oban_jobs table from growing indefinitely, Oban
provides active pruning of completed and discarded jobs.
By default, pruning retains jobs for 60 seconds.
Caveats & Guidelines
	Pruning is best-effort and performed out-of-band. This means that all limits
are soft; jobs beyond a specified age may not be pruned immediately after jobs
complete.

	Pruning is only applied to jobs that are completed or discarded (has
reached the maximum number of retries or has been manually killed). It'll
never delete a new job, a scheduled job or a job that failed and will be
retried.


Unique Jobs
The unique jobs feature lets you specify constraints to prevent enqueuing
duplicate jobs.  Uniqueness is based on a combination of args, queue,
worker, state and insertion time. It is configured at the worker or job
level using the following options:
	:period — The number of seconds until a job is no longer considered
duplicate. You should always specify a period. :infinity can be used to
indicate the job be considered a duplicate as long as jobs are retained.

	:fields — The fields to compare when evaluating uniqueness. The available
fields are :args, :queue and :worker, by default all three are used.

	:states — The job states that are checked for duplicates. The available
states are :available, :scheduled, :executing, :retryable and
:completed. By default all states are checked, which prevents any
duplicates, even if the previous job has been completed.


For example, configure a worker to be unique across all fields and states for 60
seconds:
use Oban.Worker, unique: [period: 60]
Configure the worker to be unique only by :worker and :queue:
use Oban.Worker, unique: [fields: [:queue, :worker], period: 60]
Or, configure a worker to be unique until it has executed:
use Oban.Worker, unique: [period: 300, states: [:available, :scheduled, :executing]]
Strong Guarantees
Unique jobs are guaranteed through transactional locks and database queries:
they do not rely on unique constraints in the database. This makes uniqueness
entirely configurable by application code, without the need for database
migrations.
Performance Note
If your application makes heavy use of unique jobs you may want to add an index
on the args column of the oban_jobs table. The other columns considered for
uniqueness are already covered by indexes.
Periodic Jobs
Oban allows jobs to be registered with a cron-like schedule and enqueued
automatically. Periodic jobs are registered as a list of {cron, worker} or
{cron, worker, options} tuples:
config :my_app, Oban, repo: MyApp.Repo, crontab: [
  {"* * * * *", MyApp.MinuteWorker},
  {"0 * * * *", MyApp.HourlyWorker, args: %{custom: "arg"}},
  {"0 0 * * *", MyApp.DailyWorker, max_attempts: 1},
  {"0 12 * * MON", MyApp.MondayWorker, queue: :scheduled, tags: ["mondays"]},
  {"@daily", MyApp.AnotherDailyWorker}
]
These jobs would be executed as follows:
	MyApp.MinuteWorker — Executed once every minute
	MyApp.HourlyWorker — Executed at the first minute of every hour with custom args
	MyApp.DailyWorker — Executed at midnight every day with no retries
	MyApp.MondayWorker — Executed at noon every Monday in the "scheduled" queue

The crontab format respects all standard rules and has one minute
resolution. Jobs are considered unique for most of each minute, which prevents
duplicate jobs with multiple nodes and across node restarts.
Cron Expressions
Standard Cron expressions are composed of rules specifying the minutes, hours,
days, months and weekdays. Rules for each field are comprised of literal values,
wildcards, step values or ranges:
	* — Wildcard, matches any value (0, 1, 2, ...)
	0 — Literal, matches only itself (only 0)
	*/15 — Step, matches any value that is a multiple (0, 15, 30, 45)
	0-5 — Range, matches any value within the range (0, 1, 2, 3, 4, 5)
	0-23/2 - Step values can be used in conjunction with ranges (0, 2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22)

Each part may have multiple rules, where rules are separated by a comma. The
allowed values for each field are as follows:
	minute — 0-59
	hour — 0-23
	days — 1-31
	month — 1-12 (or aliases, JAN, FEB, MAR, etc.)
	weekdays — 0-6 (or aliases, SUN, MON, TUE, etc.)

The following Cron extensions are supported:
	@hourly — 0 * * * *
	@daily (as well as @midnight) — 0 0 * * *
	@weekly — 0 0 * * 0
	@monthly — 0 0 1 * *
	@yearly (as well as @annually) — 0 0 1 1 *

Some specific examples that demonstrate the full range of expressions:
	0 * * * * — The first minute of every hour
	*/15 9-17 * * * — Every fifteen minutes during standard business hours
	0 0 * DEC * — Once a day at midnight during december
	0 7-9,4-6 13 * FRI — Once an hour during both rush hours on Friday the 13th

For more in depth information see the man documentation for cron and crontab
in your system.  Alternatively you can experiment with various expressions
online at Crontab Guru.
Caveats & Guidelines
	All schedules are evaluated as UTC unless a different timezone is configured.
See Oban.start_link/1 for information about configuring a timezone.

	Workers can be used for regular and scheduled jobs so long as they accept
different arguments.

	Duplicate jobs are prevented through transactional locks and unique
constraints. Workers that are used for regular and scheduled jobs must not
specify unique options less than 60s.

	Long running jobs may execute simultaneously if the scheduling interval is
shorter than it takes to execute the job. You can prevent overlap by passing
custom unique opts in the crontab config:


  custom_args = %{scheduled: true}

  unique_opts = [
    period: 60 * 60 * 24,
    states: [:available, :scheduled, :executing]
  ]

  config :my_app, Oban, repo: MyApp.Repo, crontab: [
    {"* * * * *", MyApp.SlowWorker, args: custom_args, unique: unique_opts},
  ]
Prioritizing Jobs
Normally, all available jobs within a queue are executed in the order they were
scheduled. You can override the normal behavior and prioritize or de-prioritize
a job by assigning a numerical priority.
	Priorities from 0-3 are allowed, where 0 is the highest priority and 3 is the
lowest.

	The default priority is 0, unless specified all jobs have an equally high
priority.

	All jobs with a higher priority will execute before any jobs with a lower
priority. Within a particular priority jobs are executed in their scheduled
order.


Testing
Oban provides some helpers to facilitate testing. The helpers handle the
boilerplate of making assertions on which jobs are enqueued. To use the
perform_job/2,3, assert_enqueued/1 and refute_enqueued/1 helpers in your
tests you must include them in your testing module and specify your app's Ecto
repo:
use Oban.Testing, repo: MyApp.Repo
Now you can assert, refute or list jobs that have been enqueued within your
integration tests:
assert_enqueued worker: MyWorker, args: %{id: 1}

# or

refute_enqueued queue: :special, args: %{id: 2}

# or

assert [%{args: %{"id" => 1}}] = all_enqueued worker: MyWorker
You can also easily unit test workers with the perform_job/2,3 function, which
automates validating job args, options, and worker results from a single
function call:
assert :ok = perform_job(MyWorker, %{id: 1})

# or

assert :ok = perform_job(MyWorker, %{id: 1}, attempt: 3, max_attempts: 3)

# or

assert {:error, _} = perform_job(MyWorker, %{bad: :arg})
See the Oban.Testing module for more details.
Caveats & Guidelines
As noted in Usage, there are some guidelines for running tests:
	Disable all job dispatching by setting queues: false or queues: nil in
your test.exs config. Keyword configuration is deep merged, so setting
queues: [] won't have any effect.

	Disable plugins via plugins: false. Default plugins, such as the fixed
pruner, aren't necessary in testing mode because jobs created within the
sandbox are rolled back at the end of the test. Additionally, the periodic
pruning queries will raise DBConnection.OwnershipError when the application
boots.

	Disable cron jobs via crontab: false. Periodic jobs aren't useful while
testing and scheduling can lead to random ownership issues.

	Be sure to use the Ecto Sandbox for testing. Oban makes use of database pubsub
events to dispatch jobs, but pubsub events never fire within a transaction.
Since sandbox tests run within a transaction no events will fire and jobs
won't be dispatched.


  config :my_app, MyApp.Repo, pool: Ecto.Adapters.SQL.Sandbox
Integration Testing
During integration testing it may be necessary to run jobs because they do work
essential for the test to complete, i.e. sending an email, processing media,
etc. You can execute all available jobs in a particular queue by calling
Oban.drain_queue/1 directly from your tests.
For example, to process all pending jobs in the "mailer" queue while testing
some business logic:
defmodule MyApp.BusinessTest do
  use MyApp.DataCase, async: true

  alias MyApp.{Business, Worker}

  test "we stay in the business of doing business" do
    :ok = Business.schedule_a_meeting(%{email: "monty@brewster.com"})

    assert %{success: 1, failure: 0} == Oban.drain_queue(:mailer)

    # Now, make an assertion about the email delivery
  end
end
See Oban.drain_queue/1 for additional details.
Error Handling
When a job returns an error value, raises an error or exits during execution the
details are recorded within the errors array on the job. When the number of
execution attempts is below the configured max_attempts limit, the job will
automatically be retried in the future.
The retry delay has an exponential backoff, meaning the job's second attempt
will be after 16s, third after 31s, fourth after 1m 36s, etc.
See the Oban.Worker documentation on "Customizing Backoff" for alternative
backoff strategies.
Error Details
Execution errors are stored as a formatted exception along with metadata about
when the failure occurred and which attempt caused it. Each error is stored with
the following keys:
	at The utc timestamp when the error occurred at
	attempt The attempt number when the error occurred
	error A formatted error message and stacktrace

See the Instrumentation docs for an example of
integrating with external error reporting systems.
Limiting Retries
By default, jobs are retried up to 20 times. The number of retries is controlled
by the max_attempts value, which can be set at the Worker or Job level. For
example, to instruct a worker to discard jobs after three failures:
use Oban.Worker, queue: :limited, max_attempts: 3
Limiting Execution Time
By default, individual jobs may execute indefinitely. If this is undesirable you
may define a timeout in milliseconds with the timeout/1 callback on your
worker module.
For example, to limit a worker's execution time to 30 seconds:
def MyApp.Worker do
  use Oban.Worker

  @impl Oban.Worker
  def perform(_args, _job) do
    something_that_may_take_a_long_time()

    :ok
  end

  @impl Oban.Worker
  def timeout(_job), do: :timer.seconds(30)
end
The timeout/1 function accepts an Oban.Job struct, so you can customize the
timeout using any job attributes.
Define the timeout value through job args:
def timeout(%_{args: %{"timeout" => timeout}}), do: timeout
Define the timeout based on the number of attempts:
def timeout(%_{attempt: attempt}), do: attempt * :timer.seconds(5)
Instrumentation and Logging
Oban provides integration with Telemetry, a dispatching library for
metrics. It is easy to report Oban metrics to any backend by attaching to
:oban events.
Here is an example of a sample unstructured log handler:
defmodule MyApp.ObanLogger do
  require Logger

  def handle_event([:oban, :job, :start], measure, meta, _) do
    Logger.warn("[Oban] :started #{meta.worker} at #{measure.system_time}")
  end

  def handle_event([:oban, :job, event], measure, meta, _) do
    Logger.warn("[Oban] #{event} #{meta.worker} ran in #{measure.duration}")
  end
end
Attach the handler to success and failure events in application.ex:
events = [[:oban, :job, :start], [:oban, :job, :stop], [:oban, :job, :exception]]

:telemetry.attach_many("oban-logger", events, &MyApp.ObanLogger.handle_event/4, [])
The Oban.Telemetry module provides a robust structured logger that handles all
of Oban's telemetry events. As in the example above, attach it within your
application.ex module:
:ok = Oban.Telemetry.attach_default_logger()
For more details on the default structured logger and information on event
metadata see docs for the Oban.Telemetry module.
Reporting Errors
Another great use of execution data is error reporting. Here is an example of
integrating with Honeybadger to report job failures:
defmodule ErrorReporter do
  def handle_event([:oban, :failure], measure, meta, _) do
    context =
      meta
      |> Map.take([:id, :args, :queue, :worker])
      |> Map.merge(measure)

    Honeybadger.notify(meta.error, context, meta.stack)
  end

  def handle_event([:oban, :trip_circuit], _measure, meta, _) do
    context = Map.take(meta, [:name])

    Honeybadger.notify(meta.error, context, meta.stack)
  end
end

:telemetry.attach_many(
  "oban-errors",
  [[:oban, :failure], [:oban, :trip_circuit]],
  &ErrorReporter.handle_event/4,
  nil
)
Isolation
Oban supports namespacing through PostgreSQL schemas, also called "prefixes" in
Ecto. With prefixes your jobs table can reside outside of your primary schema
(usually public) and you can have multiple separate job tables.
To use a prefix you first have to specify it within your migration:
defmodule MyApp.Repo.Migrations.AddPrefixedObanJobsTable do
  use Ecto.Migration

  def up do
    Oban.Migrations.up(prefix: "private")
  end

  def down do
    Oban.Migrations.down(prefix: "private")
  end
end
The migration will create the "private" schema and all tables, functions and
triggers within that schema. With the database migrated you'll then specify the
prefix in your configuration:
config :my_app, Oban,
  prefix: "private",
  repo: MyApp.Repo,
  queues: [default: 10]
Now all jobs are inserted and executed using the private.oban_jobs table. Note
that Oban.insert/2,4 will write jobs in the private.oban_jobs table, you'll
need to specify a prefix manually if you insert jobs directly through a repo.
Supervisor Isolation
Not only is the oban_jobs table isolated within the schema, but all
notification events are also isolated. That means that insert/update events will
only dispatch new jobs for their prefix. You can run multiple Oban instances
with different prefixes on the same system and have them entirely isolated,
provided you give each supervisor a distinct id.
Here we configure our application to start three Oban supervisors using the
"public", "special" and "private" prefixes, respectively:
def start(_type, _args) do
  children = [
    Repo,
    Endpoint,
    Supervisor.child_spec({Oban, name: ObanA, repo: Repo}, id: ObanA),
    Supervisor.child_spec({Oban, name: ObanB, repo: Repo, prefix: "special"}, id: ObanB),
    Supervisor.child_spec({Oban, name: ObanC, repo: Repo, prefix: "private"}, id: ObanC)
  ]

  Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
end

    Resume executing jobs in a paused queue.

      Specs

      
          option() ::
  {:circuit_backoff, timeout()}
  | {:crontab, [Oban.Config.cronjob()]}
  | {:dispatch_cooldown, pos_integer()}
  | {:name, module()}
  | {:node, binary()}
  | {:plugins, [module() | {module() | Keyword.t()}]}
  | {:poll_interval, pos_integer()}
  | {:prefix, binary()}
  | {:queues, [{atom(), pos_integer()}]}
  | {:repo, module()}
  | {:shutdown_grace_period, timeout()}
  | {:timezone, Calendar.time_zone()}
  | {:log, false | Logger.level()}

      Link to this function
    
    cancel_job(name \\ __MODULE__, job_id)

      
       
       View Source
     
      (since 1.3.0)
  


  
      Specs

      
          cancel_job(name :: atom(), job_id :: pos_integer()) :: :ok

      


Cancel an available, scheduled or retryable job and mark it as discarded to prevent it
from running. If the job is currently executing it will be killed and otherwise it is ignored.
If an executing job happens to fail before it can be cancelled the state is set to discarded.
However, if it manages to complete successfully then the state will still be completed.

  
  Example


Cancel a scheduled job with the id 1:
Oban.cancel_job(1)
:ok

Returns a specification to start this module under a supervisor.
See Supervisor.
  


    
  
    
      
      Link to this function
    
    config(name \\ __MODULE__)

      
       
       View Source
     
      (since 0.2.0)
  


  
      Specs

      
          config(name :: atom()) :: Oban.Config.t()

      


Retrieve the config struct for a named Oban supervision tree.
  


  
    
      
      Link to this function
    
    drain_queue(queue)

      
       
       View Source
     
      (since 0.4.0)
  


  
      Specs

      
          drain_queue(queue :: atom() | binary()) :: Oban.Queue.Drainer.drain_result()

      


Synchronously execute all available jobs in a queue.
See drain_queue/3.
  


  
    
      
      Link to this function
    
    drain_queue(queue, opts)

      
       
       View Source
     
      (since 0.4.0)
  


  

Synchronously execute all available jobs in a queue.
See drain_queue/3.
  


  
    
      
      Link to this function
    
    drain_queue(name, queue, opts)

      
       
       View Source
     
      (since 0.4.0)
  


  
      Specs

      
          drain_queue(name :: atom(), queue :: queue_name(), [
  Oban.Queue.Drainer.drain_option()
]) :: Oban.Queue.Drainer.drain_result()

      


Synchronously execute all available jobs in a queue.
All execution happens within the current process and it is guaranteed not to raise an error or
exit.
Draining a queue from within the current process is especially useful for testing. Jobs that are
enqueued by a process when Ecto is in sandbox mode are only visible to that process. Calling
drain_queue/3 allows you to control when the jobs are executed and to wait synchronously for
all jobs to complete.

  
  Failures & Retries


Draining a queue uses the same execution mechanism as regular job dispatch. That means that any
job failures or crashes are captured and result in a retry. Retries are scheduled in the future
with backoff and won't be retried immediately.
By default jobs are executed in safe mode, just as they are in production. Safe mode catches
any errors or exits and records the formatted error in the job's errors array.  That means
exceptions and crashes are not bubbled up to the calling process.
If you expect jobs to fail, would like to track failures, or need to check for specific errors
you can pass the with_safety: false flag.

  
  Scheduled Jobs


By default, drain_queue/3 will execute all currently available jobs. In order to execute
scheduled jobs, you may pass the :with_scheduled flag which will cause scheduled jobs to be
marked as available beforehand.

  
  Example


Drain a queue with three available jobs, two of which succeed and one of which fails:
Oban.drain_queue(:default)
%{success: 2, failure: 1}
Drain a queue including any scheduled jobs:
Oban.drain_queue(:default, with_scheduled: true)
%{success: 1, failure: 0}
Drain a queue and assert an error is raised:
assert_raise RuntimeError, fn -> Oban.drain_queue(:risky, with_safety: false) end
  


    
  
    
      
      Link to this function
    
    insert(name \\ __MODULE__, changeset)

      
       
       View Source
     
      (since 0.7.0)
  


  
      Specs

      
          insert(name :: atom(), changeset :: Ecto.Changeset.t(Oban.Job.t())) ::
  {:ok, Oban.Job.t()} | {:error, Ecto.Changeset.t()}

      


Insert a new job into the database for execution.
This and the other insert variants are the recommended way to enqueue jobs because they
support features like unique jobs.
See the section on "Unique Jobs" for more details.

  
  Example


Insert a single job:
{:ok, job} = Oban.insert(MyApp.Worker.new(%{id: 1}))
Insert a job while ensuring that it is unique within the past 30 seconds:
{:ok, job} = Oban.insert(MyApp.Worker.new(%{id: 1}, unique: [period: 30]))
  


    
  
    
      
      Link to this function
    
    insert(name \\ __MODULE__, multi, multi_name, changeset_or_fun)

      
       
       View Source
     
      (since 0.7.0)
  


  
      Specs

      
          insert(
  name :: atom(),
  multi :: Ecto.Multi.t(),
  multi_name :: Ecto.Multi.name(),
  changeset_or_fun :: Ecto.Changeset.t(Oban.Job.t()) | (... -> any())
) :: Ecto.Multi.t()

      


Put a job insert operation into an Ecto.Multi.
Like insert/2, this variant is recommended over Ecto.Multi.insert beause it supports all of
Oban's features, i.e. unique jobs.
See the section on "Unique Jobs" for more details.

  
  Example


Ecto.Multi.new()
|> Oban.insert("job-1", MyApp.Worker.new(%{id: 1}))
|> Oban.insert("job-2", fn _ -> MyApp.Worker.new(%{id: 2}) end)
|> MyApp.Repo.transaction()
  


    
  
    
      
      Link to this function
    
    insert!(name \\ __MODULE__, changeset)

      
       
       View Source
     
      (since 0.7.0)
  


  
      Specs

      
          insert!(name :: atom(), changeset :: Ecto.Changeset.t(Oban.Job.t())) ::
  Oban.Job.t()

      


Similar to insert/2, but raises an Ecto.InvalidChangesetError if the job can't be inserted.

  
  Example


job = Oban.insert!(MyApp.Worker.new(%{id: 1}))
  


    
  
    
      
      Link to this function
    
    insert_all(name \\ __MODULE__, changesets)

      
       
       View Source
     
      (since 0.9.0)
  


  
      Specs

      
          insert_all(name :: atom(), jobs :: [Ecto.Changeset.t(Oban.Job.t())]) :: [
  Oban.Job.t()
]

      


Insert multiple jobs into the database for execution.
Insertion respects prefix and log settings, but it does not use per-job unique
configuration. You must use insert/2,4 or insert!/2 for per-job unique support.
There are a few important differences between this function and Ecto.Repo.insert_all/3:
	This function always returns a list rather than a tuple of {count, records}
	This function requires a list of changesets rather than a list of maps or keyword lists


  
  Example


1..100
|> Enum.map(&MyApp.Worker.new(%{id: &1}))
|> Oban.insert_all()
  


    
  
    
      
      Link to this function
    
    insert_all(name \\ __MODULE__, multi, multi_name, changesets)

      
       
       View Source
     
      (since 0.9.0)
  


  
      Specs

      
          insert_all(
  name :: atom(),
  multi :: Ecto.Multi.t(),
  multi_name :: Ecto.Multi.name(),
  changeset :: [Ecto.Changeset.t(Oban.Job.t())]
) :: Ecto.Multi.t()

      


Put an insert_all operation into an Ecto.Multi.
This function supports the same features and has the same caveats as insert_all/2.

  
  Example


changesets = Enum.map(0..100, MyApp.Worker.new(%{id: &1}))

Ecto.Multi.new()
|> Oban.insert_all(:jobs, changesets)
|> MyApp.Repo.transaction()
  


    
  
    
      
      Link to this function
    
    pause_queue(name \\ __MODULE__, queue)

      
       
       View Source
     
      (since 0.2.0)
  


  
      Specs

      
          pause_queue(name :: atom(), queue :: queue_name()) :: :ok

      


Pause a running queue, preventing it from executing any new jobs. All running jobs will remain
running until they are finished.
When shutdown begins all queues are paused.

  
  Example


Pause the default queue:
Oban.pause_queue(:default)
:ok
  


    
  
    
      
      Link to this function
    
    resume_queue(name \\ __MODULE__, queue)

      
       
       View Source
     
      (since 0.2.0)
  


  
      Specs

      
          resume_queue(name :: atom(), queue :: queue_name()) :: :ok

      


Resume executing jobs in a paused queue.

  
  Example


Resume a paused default queue:
Oban.resume_queue(:default)
:ok
  


    
  
    
      
      Link to this function
    
    scale_queue(name \\ __MODULE__, queue, scale)

      
       
       View Source
     
      (since 0.2.0)
  


  
      Specs

      
          scale_queue(name :: atom(), queue :: queue_name(), scale :: pos_integer()) ::
  :ok

      


Scale the concurrency for a queue.

  
  Example


Scale a queue up, triggering immediate execution of queued jobs:
Oban.scale_queue(:default, 50)
:ok
Scale the queue back down, allowing executing jobs to finish:
Oban.scale_queue(:default, 5)
:ok
  


  
    
      
      Link to this function
    
    start_link(opts)

      
       
       View Source
     
      (since 0.1.0)
  


  
      Specs

      
          start_link([option()]) :: Supervisor.on_start()

      


Starts an Oban supervision tree linked to the current process.

  
  Options


These options are required; without them the supervisor won't start
	:name — used for supervisor registration, defaults to Oban
	:repo — specifies the Ecto repo used to insert and retrieve jobs


  
  Primary Options


These options determine what the system does at a high level, i.e. which queues to run.
	:crontab — a list of cron expressions that enqueue jobs on a periodic basis. See "Periodic
(CRON) Jobs" in the module docs.
For testing purposes :crontab may be set to false or nil, which disables scheduling.

	:node — used to identify the node that the supervision tree is running in. If no value is
provided it will use the node name in a distributed system, or the hostname in an isolated
node. See "Node Name" below.

	:prefix — the query prefix, or schema, to use for inserting and executing jobs. An
oban_jobs table must exist within the prefix. See the "Prefix Support" section in the module
documentation for more details.

	:queues — a keyword list where the keys are queue names and the values are the concurrency
setting. For example, setting queues to [default: 10, exports: 5] would start the queues
default and exports with a combined concurrency level of 15. The concurrency setting
specifies how many jobs each queue will run concurrently.
For testing purposes :queues may be set to false or nil, which effectively disables all
job dispatching.

	:timezone — which timezone to use when scheduling cron jobs. To use a timezone other than
the default of "Etc/UTC" you must have a timezone database like tzdata installed
and configured.

	:log — either false to disable logging or a standard log level (:error, :warn,
:info, :debug). This determines whether queries are logged or not; overriding the repo's
configured log level. Defaults to false, where no queries are logged.



  
  Twiddly Options


Additional options used to tune system behaviour. These are primarily useful for testing or
troubleshooting and don't usually need modification.
	:circuit_backoff — the number of milliseconds until queries are attempted after a database
error. All processes communicating with the database are equipped with circuit breakers and
will use this for the backoff. Defaults to 30_000ms.

	:dispatch_cooldown — the minimum number of milliseconds a producer will wait before fetching
and running more jobs. A slight cooldown period prevents a producer from flooding with
messages and thrashing the database. The cooldown period directly impacts a producer's
throughput: jobs per second for a single queue is calculated by (1000 / cooldown) * limit.
For example, with a 5ms cooldown and a queue limit of 25 a single queue can run 2,500
jobs/sec.
The default is 5ms and the minimum is 1ms, which is likely faster than the database can
return new jobs to run.

	:poll_interval - the number of milliseconds between polling for new jobs in a queue. This
is directly tied to the resolution of scheduled jobs. For example, with a poll_interval of
5_000ms, scheduled jobs are checked every 5 seconds. The default is 1_000ms.

	:shutdown_grace_period - the amount of time a queue will wait for executing jobs to complete
before hard shutdown, specified in milliseconds. The default is 15_000, or 15 seconds.



  
  Examples


To start an Oban supervisor within an application's supervision tree:
def start(_type, _args) do
  children = [MyApp.Repo, {Oban, queues: [default: 50]}]

  Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
end

  
  Node Name


When the node value hasn't been configured it is generated based on the environment:
	In a distributed system the node name is used
	In a Heroku environment the system environment's DYNO value is used
	Otherwise, the system hostname is used

  


    
  
    
      
      Link to this function
    
    start_queue(name \\ __MODULE__, opts)

      
       
       View Source
     
      (since 2.0.0)
  


  
      Specs

      
          start_queue(name :: atom(), opts :: Keyword.t()) :: :ok

      


Start a new supervised queue.
By default this starts a new supervised queue across all nodes running Oban on the same database
and prefix. You can pass the option local_only: true if you prefer to start the queue only on
the local node.

  
  Options


	queue - specifies the queue name
	limit - set the concurrency limit
	local_only - specifies if the queue will be started only on the local node, default: false


  
  Examples


Start the :priority queue with a concurrency limit of 10 across the connected nodes.
Oban.start_queue(queue: :priority, limit: 10)
:ok
Start the :media queue with a concurrency limit of 5 only on the local node.
Oban.start_queue(queue: :media, limit: 5, local_only: true)
:ok
  


    
  
    
      
      Link to this function
    
    stop_queue(name \\ __MODULE__, opts)

      
       
       View Source
     
      (since 2.0.0)
  


  
      Specs

      
          stop_queue(name :: atom(), opts :: Keyword.t()) :: :ok

      


Shutdown a queue's supervision tree and stop running jobs for that queue.
By default this action will occur across all the running nodes. Still, if you prefer to stop the
queue's supervision tree and stop running jobs for that queue only on the local node, you can
pass the option: local_only: true
The shutdown process pauses the queue first and allows current jobs to exit gracefully, provided
they finish within the shutdown limit.

  
  Options


	queue - specifies the queue name
	local_only - specifies if the queue will be stopped only on the local node, default: false


  
  Examples


Oban.stop_queue(queue: :default)
:ok

Oban.stop_queue(queue: :media, local_only: true)
:ok
  

        

      
  
    
Oban.Job    



      
A Job is an Ecto schema used for asynchronous execution.
Job changesets are created by your application code and inserted into the database for
asynchronous execution. Jobs can be inserted along with other application data as part of a
transaction, which guarantees that jobs will only be triggered from a successful transaction.

    unique_option()

    unique_state()
  



  

  
    
      Functions
    


  
    new(args, opts \\ [])
  

    Construct a new job changeset ready for insertion into the database.





  
    to_map(changeset)
  

    Convert a Job changeset into a map suitable for database insertion.

      Link to this type
    
    args()

      
       
       View Source
     
  


  
      Specs

      
          args() :: map()

      


  


  
    
      
      Link to this type
    
    errors()

      
       
       View Source
     
  


  
      Specs

      
          errors() :: [%{at: DateTime.t(), attempt: pos_integer(), error: binary()}]

      


  


  
    
      
      Link to this type
    
    option()

      
       
       View Source
     
  


  
      Specs

      
          option() ::
  {:args, args()}
  | {:max_attempts, pos_integer()}
  | {:priority, pos_integer()}
  | {:queue, atom() | binary()}
  | {:schedule_in, pos_integer()}
  | {:scheduled_at, DateTime.t()}
  | {:tags, tags()}
  | {:unique, [unique_option()]}
  | {:worker, atom() | binary()}

      


  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Oban.Job{
  __meta__: term(),
  args: args(),
  attempt: non_neg_integer(),
  attempted_at: DateTime.t(),
  attempted_by: [binary()],
  completed_at: DateTime.t(),
  discarded_at: DateTime.t(),
  errors: errors(),
  id: pos_integer(),
  inserted_at: DateTime.t(),
  max_attempts: pos_integer(),
  priority: pos_integer(),
  queue: binary(),
  scheduled_at: DateTime.t(),
  state: binary(),
  tags: tags(),
  unique: %{
    fields: [unique_field()],
    period: unique_period(),
    states: [unique_state()]
  },
  unsaved_error: %{
    kind: atom(),
    reason: term(),
    stacktrace: Exception.stacktrace()
  },
  worker: binary()
}

      


  


  
    
      
      Link to this type
    
    tags()

      
       
       View Source
     
  


  
      Specs

      
          tags() :: [binary()]

      


  


  
    
      
      Link to this type
    
    unique_field()

      
       
       View Source
     
  


  
      Specs

      
          unique_field() :: [:args | :queue | :worker]

      


  


  
    
      
      Link to this type
    
    unique_option()

      
       
       View Source
     
  


  
      Specs

      
          unique_option() ::
  {:fields, [unique_field()]}
  | {:period, unique_period()}
  | {:states, [unique_state()]}

      


  


  
    
      
      Link to this type
    
    unique_period()

      
       
       View Source
     
  


  
      Specs

      
          unique_period() :: pos_integer() | :infinity

      


  


  
    
      
      Link to this type
    
    unique_state()

      
       
       View Source
     
  


  
      Specs

      
          unique_state() :: [
  :available | :scheduled | :executing | :retryable | :completed
]

      Link to this function
    
    new(args, opts \\ [])

      
       
       View Source
     
      (since 0.1.0)
  


  
      Specs

      
          new(args(), [option()]) :: Ecto.Changeset.t()

      


Construct a new job changeset ready for insertion into the database.

  
  Options


	:max_attempts — the maximum number of times a job can be retried if there are errors
during execution
	:priority — a numerical indicator from 0 to 3 of how important this job is relative to
other jobs in the same queue. The lower the number, the higher priority the job.
	:queue — a named queue to push the job into. Jobs may be pushed into any queue, regardless
of whether jobs are currently being processed for the queue.
	:schedule_in - the number of seconds until the job should be executed
	:scheduled_at - a time in the future after which the job should be executed
	:tags — a list of tags to group and organize related jobs, i.e. to identify scheduled jobs
	:unique — a keyword list of options specifying how uniqueness will be calculated. The
options define which fields will be used, for how long, and for which states.
	:worker — a module to execute the job in. The module must implement the Oban.Worker
behaviour.


  
  Examples


Insert a job with the :default queue:
%{id: 1, user_id: 2}
|> Oban.Job.new(queue: :default, worker: MyApp.Worker)
|> Oban.insert()
Generate a pre-configured job for MyApp.Worker and push it:
%{id: 1, user_id: 2} |> MyApp.Worker.new() |> Oban.insert()
Schedule a job to run in 5 seconds:
%{id: 1} |> MyApp.Worker.new(schedule_in: 5) |> Oban.insert()
Insert a job, ensuring that it is unique within the past minute:
%{id: 1} |> MyApp.Worker.new(unique: [period: 60]) |> Oban.insert()
Insert a unique job based only on the worker field, and within multiple states:
fields = [:worker]
states = [:available, :scheduled, :executing, :retryable, :completed]

%{id: 1}
|> MyApp.Worker.new(unique: [fields: fields, period: 60, states: states])
|> Oban.insert()
  


  
    
      
      Link to this function
    
    to_map(changeset)

      
       
       View Source
     
      (since 0.9.0)
  


  
      Specs

      
          to_map(Ecto.Changeset.t(t())) :: map()

      


Convert a Job changeset into a map suitable for database insertion.

  
  Examples


Convert a worker generated changeset into a plain map:
%{id: 123}
|> MyApp.Worker.new()
|> Oban.Job.to_map()
  

        

      
  
    
Oban.Notifier    



      
The Notifier coordinates listening for and publishing notifications for events in predefined
channels.
Every Oban supervision tree contains a notifier process, registered as Oban.Notifier, which
itself maintains a single connection with an app's database. All incoming notifications are
relayed through that connection to other processes.
Channels
The notifier recognizes three predefined channels, each with a distinct responsibility:
	gossip — arbitrary communication between nodes or jobs are sent on the gossip channel
	insert — as jobs are inserted into the database an event is published on the insert
channel. Processes such as queue producers use this as a signal to dispatch new jobs.
	signal — instructions to take action, such as scale a queue or kill a running job, are sent
through the signal channel.

The insert and signal channels are primarily for internal use. Use the gossip channel to
send notifications between jobs or processes in your application.
Caveats
The notifications system is built on PostgreSQL's LISTEN/NOTIFY functionality. Notifications
are only delivered after a transaction completes and are de-duplicated before publishing.
Most applications run Ecto in sandbox mode while testing. Sandbox mode wraps each test in a
separate transaction which is rolled back after the test completes. That means the transaction
is never committed, which prevents delivering any notifications.
To test using notifications you must run Ecto without sandbox mode enabled.
Examples
Broadcasting after a job is completed:
defmodule MyApp.Worker do
  use Oban.Worker

  @impl Oban.Worker
  def perform(job) do
    :ok = MyApp.do_work(job.args)

    Oban.Notifier.notify(Oban.config(), :gossip, %{complete: job.id})

    :ok
  end
end
Listening for job complete events from another process:
def insert_and_listen(args) do
  {:ok, job} =
    args
    |> MyApp.Worker.new()
    |> Oban.insert()

  receive do
    {:notification, :gossip, %{"complete" => ^job.id}} ->
      IO.puts("Other job complete!")
  after
    30_000 ->
      IO.puts("Other job didn't finish in 30 seconds!")
  end
end

    channel()
  




  
    option()
  



  

  
    
      Functions
    


  
    child_spec(init_arg)
  

    Returns a specification to start this module under a supervisor.





  
    listen(server \\ __MODULE__, channels)
  

    Register the current process to receive relayed messages for the provided channels.





  
    notify(conf, channel, payload)
  

    Broadcast a notification to listeners on all nodes.

      Link to this type
    
    channel()

      
       
       View Source
     
  


  
      Specs

      
          channel() :: :gossip | :insert | :signal

      


  


  
    
      
      Link to this type
    
    option()

      
       
       View Source
     
  


  
      Specs

      
          option() :: {:name, module()} | {:conf, Oban.Config.t()}

      Link to this function
    
    child_spec(init_arg)

      
       
       View Source
     
  


  

Returns a specification to start this module under a supervisor.
See Supervisor.
  


    
  
    
      
      Link to this function
    
    listen(server \\ __MODULE__, channels)

      
       
       View Source
     
  


  
      Specs

      
          listen(GenServer.server(), channels :: [channel()]) :: :ok

      


Register the current process to receive relayed messages for the provided channels.
All messages are received as JSON and decoded before they are relayed to registered
processes. Each registered process receives a three element notification tuple in the following
format:
{:notification, channel :: channel(), decoded :: map()}

  
  Example


Register to listen for all :gossip channel messages:
Oban.Notifier.listen([:gossip])
Listen for messages on all channels:
Oban.Notifier.listen([:gossip, :insert, :signal])
  


  
    
      
      Link to this function
    
    notify(conf, channel, payload)

      
       
       View Source
     
  


  
      Specs

      
          notify(Oban.Config.t(), channel :: channel(), payload :: map()) :: :ok

      


Broadcast a notification to listeners on all nodes.
Notifications are scoped to the configured prefix. For example, if there are instances running
with the public and private prefixes, a notification published in the public prefix won't
be picked up by processes listening with the private prefix.

  
  Example


Broadcast a gossip message:
Oban.Notifier.notify(Oban.config(), :gossip, %{message: "hi!"})
  

        

      
  
    
Oban.Telemetry    



      
Telemetry integration for event metrics, logging and error reporting.
Job Events
Oban emits the following telemetry events for each job:
	[:oban, :job, :start] — at the point a job is fetched from the database and will execute
	[:oban, :job, :stop] — after a job succeeds and the success is recorded in the database
	[:oban, :job, :exception] — after a job fails and the failure is recorded in the database

All job events share the same details about the job that was executed. In addition, failed jobs
provide the error type, the error itself, and the stacktrace. The following chart shows which
metadata you can expect for each event:
	event	measures	metadata
	:start	:system_time	:id, :args, :queue, :worker, :attempt, :max_attempts, :prefix
	:stop	:queue_time, :duration	:id, :args, :queue, :worker, :attempt, :max_attempts, :prefix
	:exception	:queue_time, :duration	:id, :args, :queue, :worker, :attempt, :max_attempts, :prefix, :kind, :error, :stacktrace

For :exception events the metadata includes details about what caused the failure. The :kind
value is determined by how an error occurred. Here are the possible kinds:
	:error — from an {:error, error} return value. Some Erlang functions may also throw an
:error tuple, which will be reported as :error.
	:exit — from a caught process exit
	:throw — from a caught value, this doesn't necessarily mean that an error occurred and the
error value is unpredictable

Circuit Events
All processes that interact with the database have circuit breakers to prevent errors from
crashing the entire supervision tree. Processes emit a [:oban, :trip_circuit] event when a
circuit is tripped and [:oban, :open_circuit] when the breaker is subsequently opened again.
	event	metadata
	[:oban, :circuit, :trip]	:error, :message, :name, :stacktrace
	[:oban, :circuit, :open]	:name

Metadata
	:error — the error that tripped the circuit, see the error kinds breakdown above
	:name — the registered name of the process that tripped a circuit, i.e. Oban.Notifier
	:message — a formatted error message describing what went wrong
	:stacktrace — exception stacktrace, when available

Default Logger
A default log handler that emits structured JSON is provided, see attach_default_logger/0 for
usage. Otherwise, if you would prefer more control over logging or would like to instrument
events you can write your own handler.
Examples
A handler that only logs a few details about failed jobs:
defmodule MicroLogger do
  require Logger

  def handle_event([:oban, :job, :exception], %{duration: duration}, meta, nil) do
    Logger.warn("[#{meta.queue}] #{meta.worker} failed in #{duration}")
  end
end

:telemetry.attach("oban-logger", [:oban, :job, :exception], &MicroLogger.handle_event/4, nil)
Another great use of execution data is error reporting. Here is an example of integrating with
Honeybadger, but only reporting jobs that have failed 3 times or more:
defmodule ErrorReporter do
  def handle_event([:oban, :job, :exception], _, %{attempt: attempt} = meta, _) do
    if attempt >= 3 do
      context = Map.take(meta, [:id, :args, :queue, :worker])

      Honeybadger.notify(meta.error, context, meta.stack)
    end
  end
end

:telemetry.attach("oban-errors", [:oban, :job, :exception], &ErrorReporter.handle_event/4, [])

    attach_default_logger(level \\ :info)
  

    Attaches a default structured JSON Telemetry handler for logging.





  
    span(name, fun, meta \\ %{})
  

    Measure and report :start, :stop and :exception events for a function.

      Link to this function
    
    attach_default_logger(level \\ :info)

      
       
       View Source
     
      (since 0.4.0)
  


  
      Specs

      
          attach_default_logger(Logger.level()) :: :ok | {:error, :already_exists}

      


Attaches a default structured JSON Telemetry handler for logging.
This function attaches a handler that outputs logs with the following fields:
	args — a map of the job's raw arguments
	duration — the job's runtime duration, in the native time unit
	event — either :success or :failure depending on whether the job succeeded or errored
	queue — the job's queue
	source — always "oban"
	system_time — when the job started, in microseconds
	worker — the job's worker module


  
  Examples


Attach a logger at the default :info level:
:ok = Oban.Telemetry.attach_default_logger()
Attach a logger at the :debug level:
:ok = Oban.Telemetry.attach_default_logger(:debug)
  


    
  
    
      
      Link to this function
    
    span(name, fun, meta \\ %{})

      
       
       View Source
     
  


  
      Specs

      
          span(name :: atom(), fun :: (() -> term()), meta :: map()) :: term()

      


Measure and report :start, :stop and :exception events for a function.

  
  Examples


Emit span timing events for a prune function:
:ok = Oban.Telemetry.span(:prune, &MyApp.Pruner.prune/0, %{extra: :data})
That will emit the following events:
	[:oban, :prune, :start] — before the function is invoked
	[:oban, :prune, :stop] — when the function completes successfully
	[:oban, :prune, :exception] — reported if the function throws, crashes or raises an error

  

        

      
  
    
Oban.Testing    



      
This module simplifies making assertions about enqueued jobs during testing.
Assertions may be made on any property of a job, but you'll typically want to check by args,
queue or worker. If you're using namespacing through PostgreSQL schemas, also called
"prefixes" in Ecto, you should use the prefix option when doing assertions about enqueued
jobs during testing. By default the prefix option is public.
Using in Tests
The most convenient way to use Oban.Testing is to use the module:
use Oban.Testing, repo: MyApp.Repo
That will define three helper functions, assert_enqueued/1,2, refute_enqueued/1,2 and
all_enqueued/1. The functions can then be used to make assertions on the jobs that have been
inserted in the database while testing.
Some small examples:
# Assert that a job was already enqueued
assert_enqueued worker: MyWorker, args: %{id: 1}

# Assert that a job was enqueued or will be enqueued in the next 100ms
assert_enqueued [worker: MyWorker, args: %{id: 1}], 100

# Refute that a job was already enqueued
refute_enqueued queue: "special", args: %{id: 2}

# Refute that a job was already enqueued or would be enqueued in the next 100ms
refute_enqueued queue: "special", args: %{id: 2}, 100

# Make assertions on a list of all jobs matching some options
assert [%{args: %{"id" => 1}}] = all_enqueued(worker: MyWorker)
Note that the final example, using all_enqueued/1, returns a raw list of matching jobs and
does not make an assertion by itself. This makes it possible to test using pattern matching at
the expense of being more verbose.
Example
Given a simple module that enqueues a job:
defmodule MyApp.Business do
  def work(args) do
    args
    |> Oban.Job.new(worker: MyApp.Worker, queue: :special)
    |> Oban.insert!()
  end
end
The behaviour can be exercised in your test code:
defmodule MyApp.BusinessTest do
  use ExUnit.Case, async: true
  use Oban.Testing, repo: MyApp.Repo

  alias MyApp.Business

  test "jobs are enqueued with provided arguments" do
    Business.work(%{id: 1, message: "Hello!"})

    assert_enqueued worker: MyApp.Worker, args: %{id: 1, message: "Hello!"}
  end
end
Matching Scheduled Jobs and Timestamps
In order to assert a job has been scheduled at a certain time, you will need to match against
the scheduled_at attribute of the enqueued job.
in_an_hour = DateTime.add(DateTime.utc_now(), 3600, :second)
assert_enqueued worker: MyApp.Worker, scheduled_at: in_an_hour
By default, Oban will apply a 1 second delta to all timestamp fields of jobs, so that small
deviations between the actual value and the expected one are ignored. You may configure this
delta by passing a tuple of value and a delta option (in seconds) to corresponding keyword:
assert_enqueued worker: MyApp.Worker, scheduled_at: {in_an_hour, delta: 10}
Adding to Case Templates
To include helpers in all of your tests you can add it to your case template:
defmodule MyApp.DataCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      use Oban.Testing, repo: MyApp.Repo

      import Ecto
      import Ecto.Changeset
      import Ecto.Query
      import MyApp.DataCase

      alias MyApp.Repo
    end
  end
end

    all_enqueued(repo, opts)
  

    Retrieve all currently enqueued jobs matching a set of options.





  
    assert_enqueued(repo, opts)
  

    Assert that a job with particular options has been enqueued.





  
    assert_enqueued(repo, opts, timeout)
  

    Assert that a job with particular options is or will be enqueued within a timeout period.





  
    perform_job(worker, args, opts \\ [])
  

    Construct a job and execute it with a worker module.





  
    refute_enqueued(repo, opts)
  

    Refute that a job with particular options has been enqueued.





  
    refute_enqueued(repo, opts, timeout)
  

    Refute that a job with particular options is or will be enqueued within a timeout period.

      Link to this function
    
    all_enqueued(repo, opts)

      
       
       View Source
     
      (since 0.6.0)
  


  
      Specs

      
          all_enqueued(repo :: module(), opts :: Keyword.t()) :: [Oban.Job.t()]

      


Retrieve all currently enqueued jobs matching a set of options.
Only jobs matching all of the provided arguments will be returned. Additionally, jobs are
returned in descending order where the most recently enqueued job will be listed first.

  
  Examples


Assert based on only some of a job's args:
assert [%{args: %{"id" => 1}}] = all_enqueued(worker: MyWorker)
Assert that exactly one job was inserted for a queue:
assert [%Oban.Job{}] = all_enqueued(queue: :alpha)
  


  
    
      
      Link to this function
    
    assert_enqueued(repo, opts)

      
       
       View Source
     
      (since 0.3.0)
  


  
      Specs

      
          assert_enqueued(repo :: module(), opts :: Keyword.t()) :: true

      


Assert that a job with particular options has been enqueued.
Only values for the provided arguments will be checked. For example, an assertion made on
worker: "MyWorker" will match any jobs for that worker, regardless of the queue or args.
  


  
    
      
      Link to this function
    
    assert_enqueued(repo, opts, timeout)

      
       
       View Source
     
      (since 1.2.0)
  


  
      Specs

      
          assert_enqueued(repo :: module(), opts :: Keyword.t(), timeout :: pos_integer()) ::
  true

      


Assert that a job with particular options is or will be enqueued within a timeout period.
See assert_enqueued/2 for additional details.

  
  Examples


Assert that a job will be enqueued in the next 100ms:
assert_enqueued [worker: MyWorker], 100
  


    
  
    
      
      Link to this function
    
    perform_job(worker, args, opts \\ [])

      
       
       View Source
     
      (since 2.0.0)
  


  
      Specs

      
          perform_job(
  worker :: Oban.Worker.t(),
  args :: Oban.Job.args(),
  opts :: [Oban.Job.option()]
) :: Oban.Worker.result()

      


Construct a job and execute it with a worker module.
This reduces boiler plate when constructing jobs for unit tests and checks for common pitfalls.
For example, it automatically converts args to string keys before calling perform/1,
ensuring that perform clauses aren't erroneously trying to match atom keys.
The helper makes the following assertions:
	That the worker implements the Oban.Worker behaviour
	That the options provided build a valid job
	That the return is valid, e.g. :ok, {:ok, value}, {:error, value} etc.

If all of the assertions pass then the function returns the result of perform/1 for you to
make additional assertions on.

  
  Examples


Successfully execute a job with some string arguments:
assert :ok = perform_job(MyWorker, %{"id" => 1})
Successfully execute a job and assert that it returns an error tuple:
assert {:error, _} = perform_job(MyWorker, %{"bad" => "arg"})
Execute a job with the args keys automatically stringified:
assert :ok = perform_job(MyWorker, %{id: 1})
Exercise custom attempt handling within a worker by passing options:
assert :ok = perform_job(MyWorker, %{}, attempt: 42)
Cause a test failure because the provided worker isn't real:
assert :ok = perform_job(Vorker, %{"id" => 1})
  


  
    
      
      Link to this function
    
    refute_enqueued(repo, opts)

      
       
       View Source
     
      (since 0.3.0)
  


  
      Specs

      
          refute_enqueued(repo :: module(), opts :: Keyword.t()) :: false

      


Refute that a job with particular options has been enqueued.
See assert_enqueued/2 for additional details.
  


  
    
      
      Link to this function
    
    refute_enqueued(repo, opts, timeout)

      
       
       View Source
     
      (since 1.2.0)
  


  
      Specs

      
          refute_enqueued(repo :: module(), opts :: Keyword.t(), timeout :: pos_integer()) ::
  false

      


Refute that a job with particular options is or will be enqueued within a timeout period.
The minimum refute timeout is 10ms.
See assert_enqueued/2 for additional details.

  
  Examples


Refute that a job will not be enqueued in the next 100ms:
refute_enqueued [worker: MyWorker], 100
  

        

      
  
    
Oban.Worker behaviour    



      
Defines a behavior and macro to guide the creation of worker modules.
Worker modules do the work of processing a job. At a minimum they must define a perform/1
function, which is called with the full Oban.Job struct.
Defining Workers
Worker modules are defined by using Oban.Worker. A bare use Oban.Worker invocation sets a
worker with these defaults:
	:max_attempts — 20
	:priority — 0
	:queue — :default
	:unique — no uniqueness set

The following example defines a worker module to process jobs in the events queue. It then
dials down the priority from 0 to 1, limits retrying on failures to 10, adds a "business" tag,
and ensures that duplicate jobs aren't enqueued within a 30 second period:
defmodule MyApp.Workers.Business do
  use Oban.Worker,
    queue: :events,
    priority: 1,
    max_attempts: 10,
    tags: ["business"],
    unique: [period: 30]

  @impl Oban.Worker
  def perform(%Oban.Job{attempt: attempt}) when attempt > 3 do
    IO.inspect(attempt)
  end

  def perform(job) do
    IO.inspect(job.args)
  end
end
The perform/1 function receives an Oban.Job struct as argument. This allows workers to
change the behavior of perform/1 based on attributes of the job, e.g. the number of
execution attempts or when it was inserted.
The value returned from perform/1 can control whether the job is a success or a failure:
	:ok or {:ok, value} — the job is successful; for success tuples the value is ignored
	:discard — discard the job and prevent it from being retried again
	{:error, error} — the job failed, record the error and schedule a retry if possible
	{:snooze, seconds} — consider the job a success and schedule it to run seconds in the
future. Snoozing will also increase the max_attempts by one to ensure that the job isn't
accidentally discarded before it can run.

In addition to explicit return values, any unhandled exception, exit or throw will fail
the job and schedule a retry if possible.
As an example of error tuple handling, this worker will return an error tuple when the value
is less than one:
defmodule MyApp.Workers.ErrorExample do
  use Oban.Worker

  @impl Worker
  def perform(%{args: %{"value" => value}}) do
    if value > 1 do
      :ok
    else
      {:error, "invalid value given: " <> inspect(value)}
    end
  end
end
Enqueuing Jobs
All workers implement a new/2 function that converts an args map into a job changeset
suitable for inserting into the database for later execution:
%{in_the: "business", of_doing: "business"}
|> MyApp.Workers.Business.new()
|> Oban.insert()
The worker's defaults may be overridden by passing options:
%{vote_for: "none of the above"}
|> MyApp.Workers.Business.new(queue: "special", max_attempts: 5)
|> Oban.insert()
Uniqueness options may also be overridden by passing options:
%{expensive: "business"}
|> MyApp.Workers.Business.new(unique: [period: 120, fields: [:worker]])
|> Oban.insert()
Note that unique options aren't merged, they are overridden entirely.
See Oban.Job for all available options.
Customizing Backoff
When jobs fail they may be retried again in the future using a backoff algorithm. By default the
backoff is exponential with a fixed padding of 15 seconds. The default backoff is clamped to a
maximum of 24 days, the equivalent of the 20th attempt.
If the default strategy is too aggressive or otherwise unsuited to your app's workload you can
define a custom backoff function using the backoff/1 callback.
The following worker defines a backoff/1 function that delays retries using a variant of the
historic Resque/Sidekiq algorithm:
defmodule MyApp.SidekiqBackoffWorker do
  use Oban.Worker

  @impl Worker
  def backoff(%Job{attempt: attempt}) do
    trunc(:math.pow(attempt, 4) + 15 + :rand.uniform(30) * attempt)
  end

  @impl Worker
  def perform(_job) do
    :do_business
  end
end
Here are some alternative backoff strategies to consider:
	constant — delay by a fixed number of seconds, e.g. 1→15, 2→15, 3→15
	linear — delay for the same number of seconds as the current attempt, e.g. 1→1, 2→2, 3→3
	squared — delay by attempt number squared, e.g. 1→1, 2→4, 3→9
	sidekiq — delay by a base amount plus some jitter, e.g. 1→32, 2→61, 3→135

Contextual Backoff
Any error, catch or throw is temporarily recorded in the job's unsaved_error map. The unsaved
error map can be used by backoff/1 to calculate a custom backoff based on the exact error
that failed the job. In this example the backoff/1 callback checks to see if the error was
due to rate limiting and adjusts the backoff accordingly:
defmodule MyApp.ApiWorker do
  use Oban.Worker

  @five_minutes 5 * 60

  @impl Worker
  def perform(%{args: args}) do
    MyApp.make_external_api_call(args)
  end

  @impl Worker
  def backoff(%Job{attempt: attempt, unsaved_error: unsaved_error}) do
    %{kind: _, reason: reason, stacktrace: _} = unsaved_error

    case reason do
      %MyApp.ApiError{status: 429} -> @five_minutes
      _ -> trunc(:math.pow(attempt, 4))
    end
  end
end
Limiting Execution Time
By default, individual jobs may execute indefinitely. If this is undesirable you may define a
timeout in milliseconds with the timeout/1 callback on your worker module.
For example, to limit a worker's execution time to 30 seconds:
def MyApp.Worker do
  use Oban.Worker

  @impl Oban.Worker
  def perform(_job) do
    something_that_may_take_a_long_time()

    :ok
  end

  @impl Oban.Worker
  def timeout(_job), do: :timer.seconds(30)
end
The timeout/1 function accepts an Oban.Job struct, so you can customize the timeout using
any job attributes.
Define the timeout value through job args:
def timeout(%_{args: %{"timeout" => timeout}}), do: timeout
Define the timeout based on the number of attempts:
def timeout(%_{attempt: attempt}), do: attempt * :timer.seconds(5)

      Callbacks
    


  
    backoff(job)
  

    Calculate the execution backoff.





  
    new(args, opts)
  

    Build a job changeset for this worker with optional overrides.





  
    perform(job)
  

    The perform/1 function is called to execute a job.





  
    timeout(job)
  

    Set a job's maximum execution time in milliseconds.

      Link to this type
    
    result()

      
       
       View Source
     
  


  
      Specs

      
          result() ::
  :ok
  | :discard
  | {:ok, ignored :: term()}
  | {:error, reason :: term()}
  | {:snooze, seconds :: pos_integer()}

      


  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: module()

      Link to this callback
    
    backoff(job)

      
       
       View Source
     
  


  
      Specs

      
          backoff(job :: Oban.Job.t()) :: pos_integer()

      


Calculate the execution backoff.
In this context backoff specifies the number of seconds to wait before retrying a failed job.
Defaults to an exponential algorithm with a minimum delay of 15 seconds.
  


  
    
      
      Link to this callback
    
    new(args, opts)

      
       
       View Source
     
  


  
      Specs

      
          new(args :: Oban.Job.args(), opts :: [Oban.Job.option()]) :: Ecto.Changeset.t()

      


Build a job changeset for this worker with optional overrides.
See Oban.Job.new/2 for the available options.
  


  
    
      
      Link to this callback
    
    perform(job)

      
       
       View Source
     
  


  
      Specs

      
          perform(job :: Oban.Job.t()) :: result()

      


The perform/1 function is called to execute a job.
Each perform/1 function should return :ok or a success tuple. When the return is an error
tuple, an uncaught exception or a throw then the error is recorded and the job may be retried if
there are any attempts remaining.
Note that the args map provided to perform/1 will always have string keys, regardless of
the key type when the job was enqueued. The args are stored as jsonb in PostgreSQL and the
serialization process automatically stringifies all keys.
  


  
    
      
      Link to this callback
    
    timeout(job)

      
       
       View Source
     
  


  
      Specs

      
          timeout(job :: Oban.Job.t()) :: :infinity | pos_integer()

      


Set a job's maximum execution time in milliseconds.
Jobs that exceed the time limit are considered a failure and may be retried.
Defaults to :infinity.

